# Polynomial Universes and Dependent Types

C.B. Aberlé David I. Spivak

#### **Abstract**

Awodey, later with Newstead, showed how polynomial pseudomonads  $(u, 1, \Sigma)$  with extra structure (termed "natural models" by Awodey) hold within them the categorical semantics for dependent type theory. Their work presented these ideas clearly but ultimately led them outside of the category of polynomial functors in order to explain all of the structure possessed by such models of type theory.

This paper builds off that work—explicating the categorical semantics of dependent type theory by axiomatizing them *entirely* in the language of polynomial functors. In order to handle the higher-categorical coherences required for such an explanation, we work with polynomial functors internally in the language of Homotopy Type Theory, which allows for higher-dimensional structures such as pseudomonads, etc. to be expressed purely in terms of the structure of a suitably-chosen  $\infty$ -category of polynomial functors. The move from set theory to Homotopy Type Theory thus has a twofold effect of enabling a simpler exposition of natural models, which is at the same time amenable to formalization in a proof assistant, such as Agda.

Moreover, the choice to remain firmly within the setting of polynomial functors reveals many additional structures of natural models that were otherwise left implicit or not considered by Awodey & Newstead. Chief among these, we highlight the fact that every polynomial pseudomonad  $(u,1,\Sigma)$  as above that is also equipped with structure to interpret dependent product types gives rise to a self-distributive law  $u \triangleleft u \rightarrow u \triangleleft u$ , which witnesses the usual distributive law of dependent products over dependent sums.

## 1 Introduction

The central idea of dependent type theory (c.f. [Mar75]) is that *order of events* is fundamental to the mathematical story arc: when playing out any specific example story within that arc, the beginning of the story affects not only the later events, but even the very terms with which the later events will be described. For example, in the story arc of conditional probability, one may say "now if the set *P* that we are asked to condition on happens to have measure zero, we must stop; but assuming that's not the case then the result will be a new probability measure.'' Here the story teller is saying that no terms will describe what happens if *P* has measure zero, whereas otherwise the terms of standard probability will apply.

Dependent types form a logical system with syntax, rules of computation, and robust categorical semantics. In [Awo14; AN18], Awodey and later Newstead show that there is a strong connection between dependent type theory and polynomial functors, via their concept of *natural models*, which cleanly solve the problem of *strictififying* certain identities that typically hold only up to isomorphism in arbitrary categories, but must hold *strictly* in order for these to soundly model dependent type theory. The solution to this problem offered by Awodey and Newstead makes use of the type-theoretic concept of a *universe*. Such universes then turn out to naturally be regarded as polynomial functors on a suitably-chosen category of presheaves, satisfying a certain *representability* condition.

Although the elementary structure of natural models is thus straightforwardly described by considering them as objects in a category of polynomial functors, Awodey and Newstead were ultimately led outside of this category in order to fully explicate those parts of natural models that require identities to hold only *up to isomorphism*, rather than strictly. There is thus an evident tension between *strict* and *weak* identities that has not yet been fully resolved in the story of natural models. In the present work, we build on Awodey and Newstead's work to fully resolve this impasse by showing how type universes can be fully axiomatized in terms of polynomial functors, by working with polynomial functors internally in the language of *Homotopy Type Theory* (HoTT) [Voe+13]. We thus come full circle from Awodey's original motivation to develop natural models *of* Homotopy Type Theory, to describing natural models *in* Homotopy Type Theory.

The ability for us to tell the story of natural models as set entirely in the category of polynomial functors has a great simplifying effect upon the resultant theory, and reveals many additional structures, both of polynomial universes, and of the category of polynomial functors as a whole. As an illustration of this, we show how every polynomial universe u, regarded as a polynomial pseudomonad with additional structure, gives rise to self-distributive law  $u \triangleleft u \rightarrow u \triangleleft u$ , which in particular witnesses the usual distributive law of dependent products over dependent sums.<sup>1</sup>

Moreover, the move from set theory to HoTT as a setting in which to tell this story enables new tools to be applied for its telling. In particular, the account of polynomial universes we develop is well-suited to formalization in a proof assistant, and we present such a formalization in Agda. This paper is thus itself a literate Agda document in which all results have been fully formalized and checked for validity.

```
{-# OPTIONS --without-K --rewriting #-}
module poly-universes where
```

The structure of this paper is as follows:

• In Section 2, we give an introductory presentation of dependent type theory and natural models, followed by a recap the basics of HoTT that will be used throughout the rest of the paper.

 $<sup>^1</sup>$ A precursor to this story was attempted by the second-named author in a blog post, which may be useful for readers seeking intuition. There a proposed self-distributive law  $u \triangleleft u \rightarrow u \triangleleft u$  on  $u = 1 + y + y^2 + \cdots$  was claimed to witness dependent products. However, the author only later noticed that one of the four equations for distributive laws was doomed to fail [ZM22] due to the 1-dimensionality of types in u(1). The present project was inspired by this.

- In Section 3, we outline the basic theory of polynomial functors in HoTT, culminating in a demonstration of how to model dependent pair types using polynomial functors. In order to show that these polynomial functors are in fact monads, however, we will need some additional technology, to which we devote the following section.
- In Section 4, we introduce the key concept of a *polynomial universe* as a polynomial functor satisfying a certain *univalence* condition, that allows us to straightforwardly derive the monad laws for polynomial universes equipped with the structure to interpret dependent pair types.
- In Section 5, building on the ideas of the previous sections, we show how to model dependent function types with polynomial functors, and demonstrate that any polynomial universe equipped with this structure along with the aforementioned structure for interpreting dependent pair types gives rise to a self-distributive law of the corresponding monad.
- In Section 6, we conclude the paper by sketching how this theory may be further developed to handle identity types, inductive types, and other key concepts from dependent type theory.

```
open import Agda.Primitive
open import Agda.Builtin.Sigma
open import Agda.Builtin.Unit
```

# 2 Background on Type Theory, Natural Models & HoTT

We begin with a recap of natural models, dependent type theory, and HoTT, taking this also as an opportunity to introduce the basics of our Agda formalization.

# 2.1 Dependent Types and their Categorical Semantics

The question "what is a type" is as deep as it is philosophically fraught. For present purposes, however, we need not concern ourselves so much directly with what (dependent) type *are*, as with what they can *do*, and how best to mathematically model this behavior. Suffice it to say, then, that a type specifies rules for both constructing and using the *inhabitants* of that type in arbitrary contexts of usage. Following standard conventions, we use the notation a: A to mean that a is an inhabitant of type A.

In Agda, one example of such a type is the *unit type*  $\top$ , which is defined to have a single inhabitant  $\mathtt{t}\mathtt{t}$ :  $\top$ , such that for any other inhabitant  $\mathtt{x}$ :  $\top$  we must have  $\mathtt{x}$  =  $\mathtt{t}\mathtt{t}$ .

Another type (or rather, family of types) of particular importance is the *universe* of types Type, whose inhabitants themsleves represent types.<sup>2</sup> So e.g. to say that  $\top$ , as defined above, is a type, we may simply write  $\top$ : Type.

<sup>&</sup>lt;sup>2</sup>For consistency with the usage of the term "set" in HoTT (whereby sets are types satisfying a certain *truncation* condition, to be explained shortly,) we relabel Agda's universes of types as Type, rather than the default Set. We also note in passing that, due to size issues, the universe Type is not in fact one type, but

```
Type : (\ell : Level) \rightarrow Set (lsuc \ell)
Type \ell = Set \ell
```

Given a type A, one may in turn consider families of types B x indexed by, or *dependent* upon aribtrary inhabitants x : A. In agda, we represent such a type family B as a function  $A \to Type$ .

Given a type A: Type and a family of types B:  $A \rightarrow$  Type as above, two key examples of types we may construct are:

- The *dependent function type*  $(x : A) \rightarrow B x$ , whose inhabitants are functions  $\lambda x \rightarrow f x$  such that, for all a : A, we have f a : B a.
- The dependent pair type Σ A B, whose inhabitants are of the form (a , b) for a :
   A and b : B a, such that there are functions fst : Σ A B → A and snd : (p :
   Σ A B) → B (fst p).

Note that in the case where B does not depend upon x: A (i.e. the variable x does not appear in the expression for B), these correspond to the more familiar function type  $A \rightarrow B$  and pair type  $A \times B$ , respectively. E.g. we can define the Cartesian product of two types A and B as follows:

\_x\_ : 
$$\forall$$
 { $\ell$   $\kappa$ } (A : Type  $\ell$ ) (B : Type  $\kappa$ )  $\rightarrow$  Type ( $\ell$   $\sqcup$   $\kappa$ ) A × B =  $\Sigma$  A ( $\lambda$  \_  $\rightarrow$  B)

In more traditional type-theoretic notation, one might see the rules for these types written as follows:

$$\frac{\Gamma \vdash x : \top}{\Gamma \vdash \top : \mathsf{Type}} \qquad \frac{\Gamma \vdash x : \top}{\Gamma \vdash \mathsf{tt} : \top} \qquad \frac{\Gamma \vdash x : \top}{\Gamma \vdash x = \mathsf{tt}}$$
 
$$\frac{\Gamma \vdash A : \mathsf{Type} \qquad \Gamma, x : A \vdash B[x] : \mathsf{Type}}{\Gamma \vdash \Pi x : A.B[x] : \mathsf{Type}} \qquad \frac{\Gamma \vdash A : \mathsf{Type} \qquad \Gamma, x : A \vdash B[x] : \mathsf{Type}}{\Gamma \vdash \Sigma x : A.B[x] : \mathsf{Type}}$$
 
$$\frac{\Gamma, x : A \vdash f[x] : B[x]}{\Gamma \vdash \lambda x. f[x] : \Pi x : A.B[x]} \qquad \frac{\Gamma \vdash a : A \qquad \Gamma \vdash b : B[a]}{\Gamma \vdash (a,b) : \Sigma x : A.B[x]}$$
 
$$\frac{\Gamma \vdash f : \Pi x : A.B[x] \qquad \Gamma \vdash a : A}{\Gamma \vdash f a : B[a]} \qquad \frac{\Gamma \vdash p : \Sigma x : A.B[x]}{\Gamma \vdash \pi_1(p) : A} \qquad \frac{\Gamma \vdash p : \Sigma x : A.B[x]}{\Gamma \vdash \pi_2(p) : B[\pi_1(p)]}$$
 
$$(\lambda x. f[x])a = f[a] \qquad \pi_1(a,b) = a \qquad \pi_2(a,b) = b$$
 
$$f = \lambda x. fx \qquad p = (\pi_1(p), \pi_2(p))$$

The constructors  $\lambda$  and (-,-) are called the *introduction* forms of  $\Pi x: A.B[x]$  and  $\Sigma x: A.B[x]$ , while fa and  $\pi_1(p)$ ,  $\pi_2(p)$  are called the *elimination* forms of these types, respectively. One may wonder why all typing judgments in the above rules have been decorated with annotations of the form  $\Gamma \vdash$ , for some  $\Gamma$ . In these cases,  $\Gamma$  is the *context* of the corresponding judgment, used to keep track of the types of variables that may appear in that judgment.

rather a whole family of types, stratified by a hierarchy of *levels*. However, this structure of levels is not of much concern to us in this paper, so we shall do our best to ignore it.

Although contexts may seem rather trivial from a syntactic perspective, they are key to understanding the categorical semantics of dependent type theory. In particular, when modelling a dependent type theory as a category, it is the *contexts* which form the objects of this category, with morphisms between contexts being *substitutions* of terms in the domain context for the variables of the codomain context. A type A dependent upon variables in a context  $\Gamma$  is then interpreted as a morphism (i.e. substitution)  $\Gamma$ ,  $x:A\to \Gamma$ , whose domain represents the context  $\Gamma$  extended with a variable of type A. We then interpret a term a of type A in context  $\Gamma$  as a *section* of the display map representing A, i.e.

$$\Gamma \xrightarrow{a} \Gamma, x : A$$

$$\downarrow_A$$

$$\Gamma$$

Hence for each context  $\Gamma$ , there is a category  $\mathbf{Ty}[\Gamma]$ , which is the full subcategory of the slice category  $\mathcal{C}/\Gamma$  consisting of all display maps, wherein objects correspond to types in context  $\Gamma$ , and morphisms correspond to terms.

In typical categorical semantics, given a substitution  $f : \Gamma \to \Delta$ , and a type  $A : \Delta, x : A \to \Delta$ , we then interpret the action of f on A as a pullback:

$$\begin{array}{ccc}
\Gamma, x : A[f] & \longrightarrow \Delta, x : A \\
& & \downarrow_A \\
\Gamma & & \xrightarrow{f} & \Delta
\end{array}$$

In particular, then, any display map  $A : \Gamma, x : A \to \Gamma$  induces a functor  $\mathbf{Ty}[\Gamma] \to \mathbf{Ty}[\Gamma, x : A]$  by substitution along A. The left and right adjoints to this functor (if they exist) then correspond to dependent pair and dependent function types, respectively.

So far, we have told a pleasingly straightforward story of how to interpret the syntax of dependent type theory categorically. Unfortunately, this story is a fantasy, and the interpretation of type-theoretic syntax into categorical semantics sketched above is unsound, as it stands. The problem in essentials is that, in the syntax of type theory, substitution is strictly associative – i.e. given substitutions  $f:\Gamma\to\Delta$  and  $g:\Delta\to\Theta$  and a type A, we have A[g][f]=A[g[f]]; however, in the above categorical semantics, such iterated substitution is interpreted via successively taking pullbacks, which is in general only associative up to isomorphism. It seems, then, that something more is needed to account for this kind of *strictness* in the semantics of dependent type theory. It is precisely this problem which natural models exist to solve.

## 2.2 Natural Models

The key insight of Awodey [Awo14] in formulating the notion of a natural model is that the problem of strictness in the semantics of type theory has, in a sense, already been solved by the notion of *type universes*, such as Type as introduced above. Given a universe of types  $\mathcal{U}$ , rather than representing dependent types as display maps, and substitution as pullback, we can simply represent a family of types B[x] dependent

upon a type A as a function  $A \to \mathcal{U}$ , with substitution then given by precomposition, which is automatically strictly associative.

To interpret the syntax of dependent type theory in a category  $\mathcal{C}$  of contexts and substitutions, it therefore suffices to *embed*  $\mathcal{C}$  into a category whose type-theoretic internal language posesses such a universe whose types correspond to those of  $\mathcal{C}$ . For this purpose, we work in the category of *prehseaves*  $\mathbf{Set}^{\mathcal{C}^{op}}$ , with the embedding  $\mathcal{C} \hookrightarrow \mathbf{Set}^{\mathcal{C}^{op}}$  being nothing other than the Yoneda embedding.

The universe  $\mathcal U$  is then given by an object of  $\mathbf{Set}^{\mathcal C^{op}}$ , i.e. an assignment, to each context  $\Gamma$ , of a set  $\mathsf{Ty}[\Gamma]$  of types in context  $\Gamma$ , with functions  $\mathsf{Ty}[\Delta] \to \mathsf{Ty}[\Gamma]$  for each substitution  $f:\Gamma \to \Delta$  that compose associatively, together with a  $\mathcal U$ -indexed family of objects  $u \in \mathbf{Set}^{\mathcal C^{op}}/\mathcal U$ , i.e. a natural transformation  $u:\mathcal U_\bullet \to \mathcal U$ , where for each context  $\Gamma$  and type  $A \in \mathsf{Ty}[\Gamma]$ , the fibre of  $u_\Gamma$  over A is the set  $\mathsf{Tm}[\Gamma,A]$  of inhabitants of A in context  $\Gamma$ .

The condition that all types in  $\mathcal{U}$  "belong to  $\mathcal{C}$ ", in an appropriate sense, can then be expressed by requiring u to be *representable*, i.e. for any representable  $\gamma \in \mathbf{Set}^{\mathcal{C}^{op}}$  with a natural transformation  $\alpha : \gamma \Rightarrow \mathcal{U}$ , the pullback

$$\begin{array}{ccc}
\gamma \times_{\alpha,u} \mathcal{U}_{\bullet} & \longrightarrow & \mathcal{U}_{\bullet} \\
\downarrow u[\alpha] & & & \downarrow u \\
\gamma & \longrightarrow & \mathcal{U}
\end{array}$$

of u along  $\alpha$  is representable.

The question, then, is how to express that C has dependent pair types, dependent function types, etc., in terms of the structure of u. A further insight of Awodey, toward answering this question, is that u gives rise to a functor (indeed, a *polynomial functor*)  $\overline{u}: \mathbf{Set}^{C^{op}} \to \mathbf{Set}^{C^{op}}$ , defined as follows

$$\overline{u}(P)(\Gamma) = \sum_{A: \mathsf{Ty}[\Gamma]} P(\Gamma)^{\mathsf{Tm}[\Gamma, A]}$$

and much of the type-theoretic structure of u can be accounted for in terms of this functor. For instance (for reasons to be explained shortly), dependent pair types are given by a natural transformation  $\sigma: \overline{u} \circ \overline{u} \Rightarrow \overline{u}$ , that is *Cartesian* in that, for every  $\alpha: P \Rightarrow Q$ , the following naturality square is a pullback

$$\overline{u}(\overline{u}(P)) \xrightarrow{\overline{u}(\overline{u}(\alpha))} \overline{u}(\overline{u}(Q))$$

$$\sigma_P \downarrow \qquad \qquad \downarrow \sigma_Q$$

$$\overline{u}(P) \xrightarrow{\overline{u}(\alpha)} \overline{u}(Q)$$

A question that arises, then, is what structure such a natural transformation interpreting dependent pair types must possess. It is natural to think that  $\sigma$ , along with a suitably-chosen natural transformation  $Id \Rightarrow \overline{u}$ , ought to give  $\overline{u}$  the structure of a monad. However, this turns out to be too strong a requirement, as it amounts to asking that  $\Sigma x : A.(\Sigma y : B[x].C[x,y]) = \Sigma(x,y) : (\Sigma x : A.B[x]).C[x,y]$ , when in general this identity only holds up to isomorphism. Hence we seem to have crossed over from Scylla of our semantics for dependent type theory not being strict enough to interpret those

identities we expect to hold strictly, to the Charybdis of them now being too strict to interpret the identities we expect to hold only up to isomorphism. It was for this reason that Awodey & Newstead were forced to ultimately go beyond Polynomial functors in their accounts of natural models.

However, another possibility exists to solve this dilemma – to use the language of HoTT itself to reason about such equality-up-to-isomorphism in natural models. For this purpose, rather than taking natural models to be certain (representable) morphisms in  $\mathbf{Set}^{e^{op}}$ , we can instead expand the mathematical universe in which these models live to  $\mathbf{\infty}\mathbf{Grpd}^{e^{op}}$ , which, as an  $\mathbf{\infty}$ -topos, has HoTT as its internal language. Taking advantage of this fact, we can use HoTT itself as a language for studying the semantics of type theory, by postulating an abstract type  $\mathcal U$  together with a type family  $u:\mathcal U\to\mathsf{Type}$ , corresponding to a representable natural transformation  $u:\mathcal U_\bullet\to\mathcal U$  as above.

What remains, then, is to show how the various type-theoretic properties of such natural models can be expressed in terms of polynomial functors in the language of HoTT, and the complex identities to which these give rise. For this purpose, we begin with a recap of the basics of HoTT, before launching into a development of the theory of polynomial functors within HoTT, with an eye toward the latter's use in the study of natural models.

## 2.3 Homotopy Type Theory

## 2.3.1 The Identity Type

Given elements a, b: A for some type A, the identity type  $a \equiv b$  is inductively generated from the single constructor refl:  $\{x : A\} \rightarrow x \equiv x$ , witnessing reflexivity of equality.

```
open import Agda.Builtin.Equality
open import Agda.Builtin.Equality.Rewrite
```

The core insight of Homotopy Type Theory [Voe+13] is that the presence of (intensional) identity types in a system of dependent type theory endows each type with the structure of an  $\infty$ -groupoid, and endows each function between types with the structure of a functor between  $\infty$ -groupoids, etc. This allows a wealth of higher-categorical properties and structures to be defined and studied *internally* in the language of dependent type theory.

Since an invocation of reflexivity typically occurs at the end of an equality proof, we introduce the notation  $\square$  as a shorthand for refl as follows:

```
_{\square} : \forall {\ell} {A : Type \ell} (a : A) → a ≡ a a _{\square} = refl
```

The inductive generation of  $a \equiv b$  from refl then gives rise to the operation of transport that allows an inhabitant of the type B a to be converted to an inhabitant of B b for any type family B:  $(x : A) \rightarrow Type$ .

```
transp : \forall {\ell \kappa} {A : Type \ell} (B : A \rightarrow Type \kappa) {a a' : A} \rightarrow (e : a \equiv a') \rightarrow B a \rightarrow B a' transp B refl b = b
```

Transitivity of equality then follows in the usual way.<sup>3</sup>:

Symmetry of equality follows similarly:

```
sym : \forall {\ell} {A : Type \ell} {a a' : A} \rightarrow a \equiv a' \rightarrow a' \equiv a sym refl = refl
```

As mentioned above, each function  $f: A \to B$  in HoTT is canonically endowed with the structure of a functor between  $\infty$ -groupoids, where the action of such a function f on paths (i.e. elements of the identity type) is as follows:

```
ap : \forall {\ell \kappa} {A : Type \ell} {B : Type \kappa} {A a' : A} \rightarrow (f : A \rightarrow B) \rightarrow a \equiv a' \rightarrow (f a) \equiv (f a') ap f refl = refl
```

By the same token, given a proof  $f \equiv g$  for two functions  $f,g : (x : A) \rightarrow B x$ , it follows that for any a : A we have  $f a \equiv g$  a.

```
coAp : \forall {\ell \kappa} {A : Type \ell} {B : A \rightarrow Type \kappa} {f g : (x : A) \rightarrow B x} \rightarrow f \equiv g \rightarrow (x : A) \rightarrow f x \equiv g x coAp refl x = refl
```

We additionally have the following "dependent" form of ap as above, allowing us to apply a dependent function to both sides of an identity, in a suitable manner:

<sup>&</sup>lt;sup>3</sup>We also take advantage of Agda's support for mixfix notation to present transitivity in such a way as to streamline both the reading and writing of equality proofs:

```
apd : \forall {\ell0 \ell1 \kappa} {A : Type \ell0} {B : Type \ell1} {f : A \rightarrow B} \rightarrow (C : B \rightarrow Type \kappa) {a a' : A} \rightarrow (g : (x : A) \rightarrow C (f x)) \rightarrow (e : a \equiv a') \rightarrow transp C (ap f e) (g a) \equiv g a' apd B f refl = refl
```

To show that two pairs (a , b) and (a' , b') are equal, it suffices to show that there is an identification  $e: a \equiv a'$  together with  $e': transp B e b \equiv b'$ .

```
module PairEq \{\ell \ \kappa\} \{A : Type \ \ell\} \{B : A \rightarrow Type \ \kappa\} \{a \ a' : A\} \{b : B \ a\} \{b' : B \ a'\} where  pairEq : (e : a \equiv a') \ (e' : transp \ B \ e \ b \equiv b') \rightarrow (a \ , b) \equiv (a' \ , b')  pairEq refl refl = refl
```

We then have the following laws governing equality proofs for pairs.

```
pairEq\beta1 : (e : a \equiv a') (e' : transp B e b \equiv b')

\rightarrow ap fst (pairEq e e') \equiv e

pairEq\beta1 refl refl = refl

pairEq\eta : (e : (a , b) \equiv (a' , b'))

\rightarrow pairEq (ap fst e) (apd B snd e) \equiv e

pairEq\eta refl = refl
```

open PairEq public

## 2.3.2 Equivalences

A pivotal notion, both for HoTT in general, and for the content of this paper, is that of a function  $f: A \rightarrow B$  being an *equivalence* of types. The reader familiar with HoTT will know that there are several definitions – all equivalent – of this concept appearing in the HoTT literature. For present purposes, we make use of the *bi-invertible maps* notion of equivalence. Hence we say that a function  $f: A \rightarrow B$  is an equivalence if it has both a left inverse and a right inverse:

```
isEquiv : \forall \{\ell \ \kappa\} \ \{A : Type \ \ell\} \ \{B : Type \ \kappa\} \rightarrow (A \rightarrow B) \rightarrow Type \ (\ell \sqcup \kappa) isEquiv \{A = A\} \ \{B = B\} \ f =
(\Sigma \ (B \rightarrow A) \ (\lambda \ g \rightarrow (a : A) \rightarrow g \ (f \ a) \equiv a))
\times (\Sigma \ (B \rightarrow A) \ (\lambda \ h \rightarrow (b : B) \rightarrow f \ (h \ b) \equiv b))
```

Straightforwardly, the identity function at each type is an equivalence, and equivalences are closed under composition:

```
idIsEquiv : \forall \{\ell\} \{A : Type \ \ell\} \rightarrow isEquiv \{A = A\} \ (\lambda \ x \rightarrow x) idIsEquiv = ((\lambda \ x \rightarrow x) \ , \ (\lambda \ x \rightarrow refl)) \ , \ ((\lambda \ x \rightarrow x) \ , \ (\lambda \ x \rightarrow refl))
```

```
compIsEquiv : \forall {\ell0 \ell1 \ell2} {A : Type \ell0} {B : Type \ell1} {C : Type \ell2}
                   \rightarrow {g : B \rightarrow C} {f : A \rightarrow B} \rightarrow isEquiv g \rightarrow isEquiv f
                   \rightarrow isEquiv (\lambda a \rightarrow g (f a))
compIsEquiv \{g = g\} \{f = f\}
                 ((g', lg), (g'', rg))
                 ((f', lf), (f'', rf)) =
        ((\lambda c \rightarrow f'(g'c))
         , \lambda a \rightarrow (f' (g' (g (f a)))) \equiv ap f' (lg (f a)) \rangle
                    (f' (f a)
                                                ≡⟨ lf a ⟩
                                                 □)))
                    (a
      , ((\lambda c \rightarrow f'' (g'' c))
         , \lambda c \rightarrow (g (f (f'' (g'' c)))) \equiv (ap g (rf (g'' c)) \rangle
                    (g (g'' c)
                                                ≡⟨rg c ⟩
                                                  □)))
                    (c
```

A closely-related notion to equivalence is that of a function f being an *isomorphism*, i.e. having a single two-sided inverse:

```
Iso : \forall \{\ell \ \kappa\} \{A : Type \ \ell\} \{B : Type \ \kappa\} \rightarrow (A \rightarrow B) \rightarrow Type \ (\ell \sqcup \kappa)

Iso \{A = A\} \{B = B\} f =

(\Sigma (B \rightarrow A) (\lambda g \rightarrow ((a : A) \rightarrow g (f a) \equiv a)

\times ((b : B) \rightarrow f (g b) \equiv b)))
```

One might be inclined to wonder, then, why we bother to define equivalence via the seemingly more complicated notion of having both a left and a right inverse when the familiar notion of isomorphism can just as well be defined, as above. The full reasons for this are beyond the scope of this paper, though see [Voe+13] for further discussion. Suffice it to say that, for subtle reasons due to the higher-categorical structure of types in HoTT, the plain notion of isomorphism given above is not a *good* notion of equivalence, whereas that of bi-invertible maps is. In particular, the type Iso f is not necessarily a proposition for arbitrary f, whereas isEquiv f is.

We may nonetheless move more-or-less freely back and forth between the notions of equivalence and isomorphism given above, thanks to the following functions, which allow us to convert isomorphisms to equivalences and vice versa:

```
module Iso\leftrightarrowEquiv \{\ell \ \kappa\} \{A : Type \ \ell\} \{B : Type \ \kappa\} \{f : A \rightarrow B\} where

Iso\rightarrowisEquiv : Iso f \rightarrow isEquiv f
Iso\rightarrowisEquiv (g \ , 1 \ , r) = ((g \ , 1) \ , (g \ , r))

isEquiv\rightarrowIso : isEquiv f \rightarrow Iso f
isEquiv\rightarrowIso ((g \ , 1) \ , (h \ , r)) =

h \ , <math>(\lambda \ x \rightarrow (h \ (f \ x))) \equiv \langle \ \text{sym} \ (1 \ (h \ (f \ x))) \rangle

(g \ (f \ (h \ (f \ x)))) \equiv \langle \ \text{ap} \ g \ (r \ (f \ x)) \rangle
```

```
((g (f x)) \equiv \langle 1 x \rangle 
(x \square))), r
```

#### open Iso↔Equiv public

And by the above translation between equivalences and isomorphisms, each equivalence has a corresponding inverse map in the opposite direction, which is itself an equivalence:

```
module InvEquiv {ℓ κ} {A : Type ℓ} {B : Type κ} {f : A → B} where
inv : isEquiv f → B → A
inv (_ , (h , _)) = h

isoInv : (isof : Iso f) → Iso (fst isof)
isoInv (g , l , r) = (f , r , l)

invIsEquiv : (ef : isEquiv f) → isEquiv (inv ef)
invIsEquiv ef = Iso→isEquiv (isoInv (isEquiv→Iso ef))

open InvEquiv public
```

We note that, for each type family  $B: A \to Type$ , the map  $B \ a \to B \ a'$  induced by transport along  $e: a \equiv a'$  for any a, a': A is an equivalence with inverse given by transport along sym e, as follows:

## 2.3.3 Truncation, Bracket Types & Factorization

We say that a type A is:

open TranspEquiv public

- 1. *contractible* (aka (-2)-truncated) if A is uniquely inhabited
- 2. a (mere) proposition (aka (-1)-truncated) if any two elements of A are identical

3. a set (aka 0-truncated) if for any a,b : A, the type  $a \equiv b$  is a proposition.

```
isContr : \forall {\ell} \rightarrow Type \ell \rightarrow Type \ell isContr A = \Sigma A (\lambda a \rightarrow (b : A) \rightarrow a \equiv b) isProp : \forall {\ell} \rightarrow Type \ell \rightarrow Type \ell isProp A = {a b : A} \rightarrow a \equiv b isSet : \forall {\ell} \rightarrow Type \ell \rightarrow Type \ell isSet A = (a b : A) \rightarrow isProp (a \equiv b)
```

We additionally postulate the existence of a *propositional truncation*, or *bracket type* operation, that takes a type A to the least proposition (wrt entailment) entailed by inhabitation of A.

## postulate

When the type  $\| A \|$  is inhabited, we say that A is *merely* inhabited.

From this operation on types, we straightforwardly obtain the higher analogue of the usual epi-mono factorization system on functions between sets, as follows:

```
module Epi-Mono \{\ell \ \kappa\} \{A : Type \ \ell\} \{B : Type \ \kappa\} \{f : A \rightarrow B\} where
```

We say that a function  $f: A \rightarrow B$  is *injective* (i.e. a monomorphism), if for all a, b: A the map ap  $f: a \equiv b \rightarrow f$  a  $\equiv f$  b is an equivalence

```
isMono : Type (\ell \sqcup \kappa) isMono = \{a \ b : A\} \rightarrow isEquiv (ap <math>\{a = a\} \{a' = b\} f)
```

Given an element b : B, the *fibre* of f at b is the type of elements of a equipped with a proof of f  $a \equiv b$ :

```
Fibre : B \rightarrow Type (\ell \sqcup \kappa)
Fibre b = \Sigma A (\lambda a \rightarrow f a \equiv b)
```

We then say that f is *surjective* (i.e. an epimorphism), if all of its fibres are merely inhabited.

```
isEpi : Type \kappa isEpi = (b : B) \rightarrow || Fibre b || open Epi-Mono public module EMFactor \{\ell \ \kappa\} \{A : Type \ \ell\} \{B : Type \ \kappa\} (f : A \rightarrow B) where
```

Given a function f, its *image* is the type of elements of B whose fibres are merely inhabited.

```
Im : Type \kappa
Im = \Sigma B (\lambda b \rightarrow \parallel Fibre f b \parallel)
```

Every function f can then be factored into a (-1)-connected map onto its image followed by a (-1)-truncated map onto its codomain:

```
factor1 : A → Im
factor1 a = (f a), in ||-|| (a, refl)
factor2 : Im → B
factor2 (b, _) = b
factor \equiv : (a : A) \rightarrow factor 2 (factor 1 a) \equiv f a
factor \equiv a = refl
factor1IsEpi : isEpi factor1
factor1IsEpi (b , x) =
    |-||Rec ||-||IsProp
            (\lambda \{(a, refl) \rightarrow in \|-\| (a, pairEq refl \|-\|IsProp)\})
factor2IsMono : isMono factor2
factor2IsMono {a = (a , \alpha)} {b = (b , \beta)} =
    Iso→isEquiv ( (\lambda e → pairEq e \|-\|IsProp)
                    , ( (λ e → (pairEq (ap factor2 e) ||-||IsProp)
                                 \equiv \langle (ap (pairEq (ap factor2 e)) ||-|| \equiv Contr) \rangle
                                 \equiv \langle pairEq\eta e \rangle
                                 (e □)))
                      , λ e → pairEqβ1 e ||-||IsProp))
```

## open EMFactor public

Some additional facts about the identity type, that will be used in formalizing the results of this paper, are given in Appendix A.

## 3 Polynomials in HoTT

## 3.1 Basics

**Remark:** for the sake of concision, since essentially all of the categorical structures treated in this paper will be infinite-dimensional, we shall generally omit the prefix " $\infty$ -" from our descriptions these structures. Hence hereafter "category" generally means  $\infty$ -category, "functor" means  $\infty$ -functor, etc.

Let **Type** be the category of types and functions between them. Given a type **A**, let  $y^A$  denote the corresponding representable functor **Type**  $\rightarrow$  **Type**.

A *polynomial functor* is a coproduct of representable functors  $\mathbf{Type} \to \mathbf{Type}$ , i.e. an endofunctor on  $\mathbf{Type}$  of the form

$$\sum_{a:A} y^{B(a)}$$

for some type A and family of types B: A  $\rightarrow$  Type. The data of a polynomial functor is thus uniquely determined by the choice of A and B. Hence we may represent such functors in Agda simply as pairs (A, B) of this form:

Poly : 
$$(\ell \ \kappa : \text{Level}) \rightarrow \text{Type} \ ((\text{lsuc} \ \ell) \ \sqcup \ (\text{lsuc} \ \kappa))$$
  
Poly  $\ell \ \kappa = \Sigma \ (\text{Type} \ \ell) \ (\lambda \ A \rightarrow A \rightarrow \text{Type} \ \kappa)$ 

A basic example of such a polynomial functor is the identity functor y consisting of a single term of unit arity – hence represented by the pair ( $\top$ ,  $\lambda \rightarrow \top$ ).

$$y$$
 : Poly lzero lzero  $y$  = ( $\top$  ,  $\lambda$   $\_$   $\rightarrow$   $\top$ )

The observant reader may note the striking similarity of the above-given formula for a polynomial functor and the endofunctor on  $\mathbf{Set}^{^{C^{op}}}$  defined in the previous section on natural models. Indeed, this is no accident, given a type U and a function  $u:U \to T$  Type corresponding to a natural model as described previously, we obtain the corresponding polynomial u:P oly as the pair (U,u). Hence we can study the structure of U and u in terms of u, and this, as we shall see, allows for some significant simplifications in the theory of natural models.

Given polynomial functors  $p = \sum_{a:A} y^{B(a)}$  and  $q = \sum_{c:C} y^{D(c)}$ , a natural transformation  $p \Rightarrow q$  is equivalently given by the data of a *forward* map  $f : A \rightarrow B$  and a *backward* map  $g : (a : A) \rightarrow D$  ( $f a) \rightarrow B$  a, as can be seen from the following argument via Yoneda:

$$\int_{y \in \mathbf{Type}} \left( \sum_{a:A} y^{B(a)} \right) \to \sum_{c:C} y^{D(c)}$$

$$\simeq \prod_{a:A} \int_{y \in \mathbf{Type}} y^{B(a)} \to \sum_{c:C} y^{D(c)}$$

$$\simeq \prod_{a:A} \sum_{c:C} B(a)^{D(c)}$$

$$\simeq \sum_{f:A \to C} \prod_{a:A} B(a)^{D(f(c))}$$

We use the notation  $p \subseteq q$  to denote the type of natural transformations from p to q (aka *lenses* from p to q), which may be written in Agda as follows:

```
\_\leftrightarrows\_ : \forall {\ell0 \ell1 \kappa0 \kappa1} \rightarrow Poly \ell0 \kappa0 \rightarrow Poly \ell1 \kappa1 \rightarrow Type (\ell0 \sqcup \ell1 \sqcup \kappa0 \sqcup \kappa1) (A , B) \leftrightarrows (C , D) = \Sigma (A \rightarrow C) (\lambda f \rightarrow (a : A) \rightarrow D (f a) \rightarrow B a)
```

By application of function extensionality, we derive the following type for proofs of equality between lenses:

For each polynomial *p*, the corresponding identity lens is given by the following data:

```
id : \forall \{\ell \ \kappa\} \ (p : Poly \ \ell \ \kappa) \rightarrow p \Longrightarrow p
id p = ((\lambda \ a \rightarrow a), \lambda \ a \ b \rightarrow b)
```

And given lenses  $p \subseteq q$  and  $q \subseteq r$ , their composition may be computed as follows:

```
comp : \forall {\ell 0 \ \ell 1 \ \ell 2 \ \kappa 0 \ \kappa 1 \ \kappa 2} 
 \rightarrow {p : Poly \ell 0 \ \kappa 0} {q : Poly \ell 1 \ \kappa 1} (r : Poly \ell 2 \ \kappa 2) 
 \rightarrow p \leftrightarrows q \rightarrow q \leftrightarrows r \rightarrow p \leftrightarrows r 
 comp r (f , f\sharp) (g , g\sharp) = 
 ( (\lambda a \rightarrow g (f a)) , \lambda a z \rightarrow f\sharp a (g\sharp (f a) z) )
```

Hence we have a category **Poly** of polynomial functors and lenses between them. Our goal, then, is to show how the type-theoretic structure of a natural model naturally arises from the structure of this category. In fact, **Poly** is replete with categorical structures of all kinds, of which we now mention but a few:

# 3.2 The Vertical-Cartesian Factorization System on Poly

We say that a lens  $(f, f\sharp)$ :  $(A, B) \hookrightarrow (C, D)$  is *vertical* if  $f: A \rightarrow C$  is an equivalence, and Cartesian if for every a: A, the map  $f\sharp a: D[f a] \rightarrow B$  a is an equivalence.<sup>4</sup>

```
module Vert-Cart \{\ell 0 \ \ell 1 \ \kappa 0 \ \kappa 1\} \ \{p : Poly \ \ell 0 \ \kappa 0\} (q : Poly \ \ell 1 \ \kappa 1) \ (f : p \leftrightarrows q) \ \text{where} is Vertical : \underbrace{Set}_{} (\ell 0 \ \sqcup \ \ell 1) is Vertical = is Equiv \ (fst \ f)
```

<sup>&</sup>lt;sup>4</sup>For a proof that this notion of Cartesian morphism between polynomials is equivalent to the one given previously in Section 2.2, see Chapter 5.5 of [NS24]

```
isCartesian : Set (\ell 0 \sqcup \kappa 0 \sqcup \kappa 1) isCartesian = (a : fst p) \rightarrow isEquiv (snd f a) open Vert-Cart public
```

Every lens (A , B)  $\leftrightarrows$  (C , D) can then be factored as a vertical lens followed by a Cartesian lens:

```
module VertCartFactor {$\ell 0 \ \ell 1 \ \kappa 0 \ \kappa 1$} {p} : Poly $\ell 0 \ \kappa 0$} (q: Poly $\ell 1 \ \kappa 1) (f: p \leftrightarrows q)$ where  
    vcIm : Poly $\ell 0 \ \kappa 1$ vcIm = (fst p , $\lambda \ x \to snd q (fst f x))  
    vertfactor : p \leftrightarrows vcIm vertfactor = ( ($\lambda \ x \to x) \ , ($\lambda \ a \ x \to snd f \ a \ x) )  
    vertfactorIsVert : isVertical vcIm vertfactor vertfactorIsVert = idIsEquiv  
    cartfactor : vcIm \leftrightarrows q  
    cartfactor = ( fst f , ($\lambda \ a \ x \to x) )  
    cartfactorIsCart : isCartesian q cartfactor cartfactorIsCart x = idIsEquiv  
    vertcartfactor = : EqLens q f (comp q vertfactor cartfactor)  
    vertcartfactor = = ( ($\lambda \ a \ \tau \ ref1) \ , ($\lambda \ a \ b \ \tau \ ref1) )
```

open VertCartFactor public

Of these two classes of morphisms in **Poly**, it is *Cartesian* lenses that shall be of principal interest to us. If we view a polynomial p = (A, B) as an A-indexed family of types, given by B, then given a lens  $(f, f\sharp): p \subseteq u$ , we can think of the map  $f\sharp a: u(fa) \to B$  a, as an *elimination form* for the type u(fa), i.e. a way of *using* elements of the type u(fa). If we then ask that  $(f, f\sharp)$  is Cartesian, this implies that the type u(fa) is completely characterized (up to equivalence) by this elimination form, and in this sense, *u contains* the type u(fa) and u(fa) and u(fa) is completely characterized (up to equivalence) by this elimination form, and in this sense, *u contains* the type u(fa) and u(fa) and u(fa) and u(fa) is completely characterized (up to equivalence) by this elimination form, and in this sense, *u contains* the type u(fa) and u(fa) are the same type u(fa) are the same type u(fa) and u(fa) are the same type u(fa) and u(fa) are the same type u(fa) and u(fa) are the same type u(fa) and u(fa) are the same type u(fa) are the same type

We can therefore use Cartesian lenses to detect which types are contained in a natural model u.

<sup>&</sup>lt;sup>5</sup>Those familiar with type theory may recognize this practice of defining types in terms of their elimination forms as characteristic of so-called *negative* types (in opposition to *positive* types, which are characterized by their introduction forms). Indeed, there are good reasons for this, having to do with the fact that negative types are equivalently those whose universal property is given by a *representable* functor, rather than a *co-representable* functor, which reflects the fact that natural models are defined in terms of *presheaves* on a category of contexts, rather than *co-presheaves*.

A further fact about Cartesian lenses is that they are closed under identity and composition, as a direct consequence of the closure of equivalences under identity and composition:

Hence there is a category **Poly**<sup>Cart</sup> defined as the wide subcategory of **Poly** whose morphisms are precisely Cartesian lenses. As we shall see, much of the categorical structure of natural models qua polynomial functors can be derived from the subtle interplay between **Poly**<sup>Cart</sup> and **Poly**.

## 3.2.1 Epi-Mono Factorization on Poly<sup>Cart</sup>

In fact, **Poly**<sup>Cart</sup> itself inherits a factorization system from the epi-mono factorization on types considered previously.

Define a lens (f , f $\sharp$ ) : p  $\leftrightarrows$  q to be a *vertical embedding* if f is a monomorphism, and a *vertical surjection* if f is an epimorphism.

```
module VertEpi-Mono \{\ell 0 \ \ell 1 \ \kappa 0 \ \kappa 1\} \ \{p : Poly \ \ell 0 \ \kappa 0\}  (q : Poly \ \ell 1 \ \kappa 1) \ (f : p \leftrightarrows q) where is Vertical Embedding : Set (\ell 0 \sqcup \ell 1) is Vertical Embedding = is Mono (fst f) is Vertical Surjection : Set \ell 1 is Vertical Surjection = is Epi (fst f) open VertEpi-Mono public
```

Then every Cartesian lens can be factored into a vertical surjection followed by a vertical embedding, both of which are Cartesian.

```
module CartEMFactorization {$\ell 0$ $\ell 1$ $\kappa 0$ $\kappa 1$} {p: Poly $\ell 0$ $\kappa 0$} $$ (q: Poly $\ell 1$ $\kappa 1$) (f: p $\leftrightarrows q$) (cf: isCartesian q f) where $$ cartIm: Poly $\ell 1$ $\kappa 1$
```

```
factorcart1 : p \subseteq cartIm
     factorcart1 = ( factor1 (fst f) , snd f )
     factorcart1IsCart : isCartesian cartIm factorcart1
     factorcart1IsCart = cf
     factorcart1IsEpi : isVerticalSurjection cartIm factorcart1
     factorcart1IsEpi = factor1IsEpi (fst f)
     factorcart2 : cartIm \leftrightarrows q
     factorcart2 = ( factor2 (fst f) , (\lambda - y \rightarrow y) )
     factorcart2IsCart : isCartesian q factorcart2
     factorcart2IsCart _ = idIsEquiv
     factorcart2IsMono : isVerticalEmbedding q factorcart2
     factorcart2IsMono = factor2IsMono (fst f)
     factorcart≡ : EqLens q f (comp q factorcart1 factorcart2)
     factorcart = = ( (\lambda x \rightarrow refl) , (\lambda x y \rightarrow refl) )
open CartEMFactorization public
   We note in passing that the vertical embeddings are indeed the monomorphisms in
Poly<sup>Cart</sup>, i.e. if f : q \subseteq r is a both Cartesian and a vertical embedding, then for any
Cartesian g h : p \leftrightarrows q such that f \circ g \equiv f \circ h, we have g = h.
VertEmbedding→PolyCartMono :
     \forall \{\ell 0 \ \ell 1 \ \ell 2 \ \kappa 0 \ \kappa 1 \ \kappa 2\} \{p : Poly \ \ell 0 \ \kappa 0\}
     \rightarrow {q : Poly \ell 1 \kappa 1} (r : Poly \ell 2 \kappa 2) {f : q \rightleftharpoons r}
     \rightarrow isCartesian r f \rightarrow isVerticalEmbedding r f
     \rightarrow {g h : p \leftrightarrows q} \rightarrow isCartesian q g \rightarrow isCartesian q h
     \rightarrow EqLens r (comp r g f) (comp r h f)
     → EqLens q g h
VertEmbedding→PolyCartMono \{p = p\} \{q = q\} r \{f = (f, f\sharp)\} cf vef
                                     {g = (g, g\sharp)} {h = (h, h\sharp)} cg ch (e, e\sharp) =
     ( (\lambda a \rightarrow inv vef (e a))
     , (\lambda \text{ a d} \rightarrow (g \sharp \text{ a d}))
                          \equiv \langle ap (g \sharp a) (sym (snd (snd (cf (g a))) d)) \rangle
                    ( \_ \equiv \langle (e \sharp a (inv (cf (g a)) d)) \rangle
                    ( \_ \equiv \langle (ap (h \sharp a))
                                     ( \_ \equiv \langle (ap (f \sharp (h a))) \rangle
                                                      (transpPre vef
```

cartIm = (Im (fst f) ,  $\lambda$  (x , \_)  $\rightarrow$  snd q x)

```
(\lambda \ x \ y \rightarrow inv \ (cf \ x) \ y)
(e \ a))) \ \rangle
(\ \_ \equiv \langle \ snd \ (snd \ (cf \ (h \ a))) \ \_ \rangle
(\ \_ \square)))) \ \rangle
((h\sharp \ a \ (transp \ (snd \ q) \ (inv \ vef \ (e \ a)) \ d)) \ \square)))) )
```

## 3.3 Composition of Polynomial Functors

As endofunctors on **Type**, polynomial functors may straightforwardly be composed. To show that the resulting composite is itself (equivalent to) a polynomial functor, we can reason via the following chain of equivalences: given polynomials (A , B) and (C , D), their composite, evaluated at a type y is

$$\begin{array}{ll} & \sum_{a:A} \prod_{b:B(a)} \sum_{c:C} y^{D(c)} \\ \simeq & \sum_{a:A} \sum_{f:B(a) \to C} \prod_{b:B(a)} y^{D(f(b))} \\ \simeq & \sum_{(a,f):\sum_{a:A} C^{B(a)}} y^{\sum_{b:B(a)} D(f(b))} \end{array}$$

This then defines a monoidal product  $\triangleleft$  on **Poly** with monoidal unit given by the identity functor y:

where ∢Lens is the action of ∢ on lenses.

By construction, the existence of a Cartesian lens ( $\sigma$  ,  $\sigma\sharp$ ) :  $u \triangleleft u \leftrightarrows u$  effectively shows that u is closed under  $\Sigma$ -types, since:

•  $\sigma$  maps a pair (A, B) consisting of A: U and B: (u A)  $\rightarrow$  U to a term  $\sigma$  (A, B) representing the  $\Sigma$  type. This corresponds to the type formation rule

$$\frac{\Gamma \vdash A : \mathsf{Type} \qquad \Gamma, x : A \vdash B[x] \; \mathsf{Type}}{\Gamma \vdash \Sigma x : A.B[x] \; \mathsf{Type}}$$

• For all (A , B) as above,  $\sigma \sharp$  (A , B) takes a term of type  $\sigma$  (A , B) and yields a term fst ( $\sigma \sharp$  (A , B)) : A along with a term snd ( $\sigma \sharp$  (A , B)) : B (fst

 $(\sigma \sharp (A, B))$ , corresponding to the elimination rules

$$\frac{\Gamma \vdash p : \Sigma x : A.B[x]}{\Gamma \vdash \pi_1(p) : A} \quad \frac{\Gamma \vdash p : \Sigma x : A.B[x]}{\Gamma \vdash \pi_2(p) : B[\pi_1(p)]}$$

• The fact that  $\sigma \sharp$  (A , B) has is an equivalence implies it has an inverse  $\sigma \sharp^{-1}$  (A , B) :  $\Sigma$  (u A) ( $\lambda$  x  $\rightarrow$  u (B x))  $\rightarrow$  u ( $\sigma$  (A , B)), which takes a pair of terms to a term of the corresponding pair type, and thus corresponds to the introduction rule

$$\frac{\Gamma \vdash a : A \qquad \Gamma \vdash b : B[a]}{\Gamma \vdash (a,b) : \Sigma x : A.B[x]}$$

• The fact that  $\sigma^{\sharp^{-1}}(A, B)$  is both a left and a right inverse to  $\sigma^{\sharp}$  then implies the usual  $\beta$  and  $\eta$  laws for dependent pair types

$$\pi_1(a,b) = a \quad \pi_2(a,b) = b \quad p = (\pi_1(p), \pi_2(p))$$

Similarly, the existence of a Cartesian lens  $(\eta, \eta \sharp)$  :  $y \leftrightharpoons u$  implies that u contains (a type equivalent to) the unit type, in that:

 There is an element η tt: *U* which represents the unit type. This corresponds to the type formation rule

$$\Gamma \vdash \top : \mathsf{Type}$$

- The "elimination rule"  $\eta \sharp$  tt :  $\mathfrak{u}(\eta \ \mathsf{tt}) \to \top$ , applied to an element  $\mathfrak{x} : \mathfrak{u}(\eta \ \mathsf{tt})$  is trivial, in that it simply discards  $\mathfrak{x}$ . This corresponds to the fact that, in the ordinary type-theoretic presentation,  $\top$  does not have an elimination rule.
- However, since this trivial elimination rule has an inverse  $\eta^{\sharp^{-1}}$  tt :  $\top \to u$  ( $\eta$  tt), it follows that there is a (unique) element  $\eta^{\sharp^{-1}}$  tt tt : u ( $\eta$  tt), which corresponds to the introduction rule for  $\top$ :

$$\Gamma$$
 + tt :  $\top$ 

• Moreover, the uniqueness of this element corresponds to the  $\eta$ -law for  $\top$ :

$$\frac{\Gamma \vdash x : \top}{\Gamma \vdash x = \mathsf{tt}}$$

But then, what sorts of laws can we expect Cartesian lenses as above to obey, and is the existence of such a lens all that is needed to ensure that the natural model u has dependent pair types in the original sense of Awodey & Newstead's definition in terms of Cartesian (pseudo)monads [Awo14; AN18], or is some further data required? And what about  $\Pi$  types, or other type formers? To answer these questions, we will need to study the structure of  $\triangleleft$ , along with some closely related functors, in a bit more detail. In particular, we shall see that the structure of  $\triangleleft$  as a monoidal product on **Poly** reflects many of the basic identities one expects to hold of  $\Sigma$  types.

For instance, the associativity of  $\triangleleft$  corresponds to the associativity of  $\Sigma$ -types.

### open ⊲Assoc public

while the left and right unitors of  $\triangleleft$  correspond to the fact that  $\top$  is both a left and a right unit for  $\Sigma$ -types.

 $\forall unitr^{-1} = ((\lambda a \rightarrow a, (\lambda \rightarrow tt)), (\lambda a (b, ) \rightarrow b))$ 

module  $\triangleleft$ LRUnit  $\{\ell \ \kappa\}$  (p : Poly  $\ell \ \kappa$ ) where

## open ⊲LRUnit public

 $\triangleleft$ unitr<sup>-1</sup> : p  $\leftrightarrows$  (p  $\triangleleft$  y)

n fact, ⊲ restricts to a monoidal product on **Poly**<sup>Cart</sup>, since the functorial action of ⊲ on lenses preserves Cartesian lenses,

and all of the above-defined structure morphisms for  $\triangleleft$  are Cartesian.

```
module \triangleleftAssocCart {ℓ0 ℓ1 ℓ2 κ0 κ1 κ2} (p : Poly ℓ0 κ0)
                            (q : Poly \ell 1 \kappa 1) (r : Poly \ell 2 \kappa 2) where
      \triangleleftassocCart : isCartesian (p \triangleleft (q \triangleleft r)) (\triangleleftassoc p q r)
      ⊲assocCart _ =
            Iso→isEquiv
                   ( snd (\triangleleftassoc<sup>-1</sup> p q r) _
                   , ( (\lambda \rightarrow refl) , (\lambda \rightarrow refl) ) )
      \triangleleftassoc<sup>-1</sup>Cart : isCartesian ((p \triangleleft q) \triangleleft r) (\triangleleftassoc<sup>-1</sup> p q r)
      ⊲assoc<sup>-1</sup>Cart _ =
            Iso→isEquiv
                   ( snd (\triangleleftassoc p q r) _
                   , ( (\lambda \rightarrow refl) , (\lambda \rightarrow refl) )
open ⊲AssocCart public
module \triangleleftLRUnitCart {ℓ κ} (p : Poly ℓ κ) where
      ⊲unitlCart : isCartesian p (⊲unitl p)
      dunitlCart _ =
            Iso→isEquiv
                   ( snd ( unitl^{-1} p) _{-}
                   , ((\lambda \rightarrow refl), (\lambda \rightarrow refl))
      \neg unitl^{-1}Cart : isCartesian (y \triangleleft p) (\neg unitl^{-1} p)
      dunitl<sup>-1</sup>Cart _ =
            Iso→isEquiv
                   ( snd (⊲unitl p) _
                   , ((\lambda \rightarrow refl), (\lambda \rightarrow refl))
      dunitrCart : isCartesian p (dunitr p)
      ⊲unitrCart _ =
            Iso→isEquiv
                   ( snd (\triangleleftunitr^{-1} p) _
                   , ((\lambda \_ \rightarrow refl) , (\lambda \_ \rightarrow refl)) )
      \triangleleftunitr<sup>-1</sup>Cart : isCartesian (p \triangleleft y) (\triangleleftunitr<sup>-1</sup> p)
      dunitr<sup>-1</sup>Cart _ =
            Iso→isEquiv
                   ( snd (⊲unitr p) _
                   , ((\lambda \rightarrow refl), (\lambda \rightarrow refl))
```

#### open ⊲LRUnitCart public

We should expect, then, for these equivalences to be somehow reflected in the structure of a Cartesian lenses  $\eta: y \hookrightarrow u$  and  $\mu: u \triangleleft u \hookrightarrow u$ . This would be the case, e.g., if the following diagrams in **Poly**<sup>Cart</sup> were to commute

$$y \triangleleft \mathfrak{U} \xrightarrow{\eta \triangleleft \mathfrak{U}} \mathfrak{U} \triangleleft \mathfrak{U} \xleftarrow{\mathfrak{U} \triangleleft \eta} \mathfrak{U} \triangleleft y \qquad \qquad (\mathfrak{U} \triangleleft \mathfrak{U}) \triangleleft \mathfrak{U} \xrightarrow{\operatorname{dassoc}} \mathfrak{U} \triangleleft (\mathfrak{U} \triangleleft \mathfrak{U}) \xrightarrow{\mathfrak{U} \triangleleft \mu} \mathfrak{U} \triangleleft \mathfrak{U} \qquad \qquad \downarrow \mu \qquad \qquad$$

One may recognize these as the usual diagrams for a monoid in a monoidal category, hence (since  $\triangleleft$  corresponds to composition of polynomial endofunctors) for a *monad* as typically defined. However, because of the higher-categorical structure of types in HoTT, we should not only ask for these diagrams to commute, but for the cells exhibiting that these diagrams commute to themselves be subject to higher coherences, and so on, giving u not the structure of a (Cartesian) monad, but rather of a (Cartesian)  $\infty$ -*monad*.

Yet demonstrating that u is an  $\infty$ -monad involves specifying a potentially infinite amount of coherence data. Have we therefore traded both the Scylla of equality-up-to-isomorphism and the Charybdis of strictness for an even worse fate of higher coherence hell? The answer to this question, surprisingly, is negative, as there is a way to implicitly derive all of this data from a single axiom, which corresponds to the characteristic axiom of HoTT itself: univalence. To show this, we now introduce the central concept of this paper – that of a *polynomial universe*.

# 4 Polynomial Universes

## 4.1 Univalence

For any polynomial u = (U , E1), we say that u is univalent if u is a subterminal object in  $Poly^{Cart}$ , i.e. for any other polynomial p, the type of Cartesian lenses  $p \subseteq u$  is a proposition, i.e. any two Cartesian lenses with codomain u are equal.

```
isUnivalent : \forall {\ell \kappa} → Poly \ell \kappa → Set\omega isUnivalent u = \forall {\ell' \kappa'} {p : Poly \ell' \kappa'} \rightarrow {f g : p \leftrightarrows u} \rightarrow isCartesian u f \rightarrow isCartesian u g \rightarrow EqLens u f g
```

We call this property of polynomials univalence in analogy with the usual univalence axiom of HoTT. Indeed, the univalence axiom can be equivalently stated as the fact that the polynomial functor (Type ,  $\lambda$  X  $\rightarrow$  X) is itself univalent.

### postulate

```
ua : \forall \{\ell\} \rightarrow \text{isUnivalent (Type } \ell \text{ , } \lambda \text{ X } \rightarrow \text{X})
```

We shall refer to univalent polynomial functors as *polynomial universes*. f we think of a polynomial p as representing a family of types, then what this tells us is that if u is a polynomial universe, there is essentially at most one way for u to contain the types represented by p, where Containment is here understood as existence of a Cartesian lens  $p \subseteq u$ . In this case, we say that u *classifies* the types represented by p.

As a direct consequence of this fact, it follows that every diagram consisting of parallel Cartesian lenses into a polynomial universe automatically commutes, and moreover, every higher diagram that can be formed between the cells exhibiting such commutation must also commute, etc.

Hence the fact that u must satisfy the laws of a monad if there are Cartesian lenses  $\eta$ :  $y \subseteq u$  and  $\mu$ :  $u \triangleleft u \subseteq u$  follows immediately from the above theorem and the closure of Cartesian lenses under composition:

```
module UnivMonad \{\ell \ \kappa\} (u : Poly \ \ell \ \kappa) (univ : isUnivalent u)
                           (\eta : y \leftrightarrows u) (\mu : (u \triangleleft u) \leftrightarrows u)
                           (c\eta : isCartesian u \eta) (c\mu : isCartesian u \mu) where
      univ\triangleleftunitl : EqLens u (\triangleleftunitl u) (comp u (\eta \triangleleft \triangleleft [u] (id u)) \mu)
      univ⊲unitl = univ (⊲unitlCart u)
                                  (compCartesian u (\triangleleftCart u u c\eta (idCart u)) c\mu)
      univ\triangleleftunitr : EqLens u (\triangleleftunitr u) (comp u ((id u) \triangleleft \triangleleft[ u ] \eta) \mu)
      univ⊲unitr = univ (⊲unitrCart u)
                                  (compCartesian u (\triangleleftCart u u (idCart u) c\eta) c\mu)
      univassoc : EqLens u (comp u (\mu \triangleleft [u] (id u)) \mu)
                                         (comp \ u \ (\neg assoc \ u \ u \ u)
                                                     (comp \ u \ ((id \ u) \ \triangleleft \neg [\ u \ ] \ \mu) \ \mu))
      univ\triangleleftassoc = univ (compCartesian u (\triangleleftCart u u c\mu (idCart u)) c\mu)
                                  (compCartesian u (\triangleleftassocCart u u u)
                                                             (compCartesian u
                                                                (\triangleleft \mathsf{Cart} \ u \ u \ (\mathsf{idCart} \ u) \ \mathsf{c}\mu) \ \mathsf{c}\mu))
```

## open UnivMonad public

And more generally, all the higher coherences of an  $\infty$ -monad would follow – if we bothered to write them out – from the contractibility of the types of Cartesian lenses p  $\leftrightarrows u$  that can be formed using  $\mu$  and  $\eta$ .

## 4.1.1 Examples of Polynomial Universes

We have so far seen that polynomial universes are quite special objects in the theory of polynomial functors in HoTT, but what good would such special objects do us if they turned out to be exceedingly rare or difficult to construct?

In fact, polynomial universes are surprisingly plentiful in univalent type theory. We have already seen how the univalence axiom implies that (Type ,  $\lambda$  X  $\rightarrow$  X) is a polynomial universe. From this single example, a plethora of others can be seen to follow, many of which encompass familiar constructs from programming and mathematics.

In a sense, the polynomial (Type ,  $\lambda$  X  $\rightarrow$  X) is *universal* among polynomials in  $\mathbf{Poly}^{\mathbf{Cart}}$  in that, for any polynomial p, there is a (necessarily unique, by univalence) Cartesian morphism p  $\leftrightarrows$  (Type ,  $\lambda$  X  $\rightarrow$  X). Or rather, there would be, were it not for the size issues preventing (Type ,  $\lambda$  X  $\rightarrow$  X) from being a single object. Instead, it can more accurately be said that the family of polynomials (Type  $\ell$  ,  $\lambda$  X  $\rightarrow$  X) for all  $\ell$ : Level is universal among polynomials in  $\mathbf{Poly}^{\mathbf{Cart}}$  – this can be shown straightforwardly as follows:

```
module PolyCartUniv \{\ell \ \kappa\} (p : Poly \ell \ \kappa) where classifier : p \leftrightarrows (Type \kappa , \lambda X \rightarrow X) classifier = (snd p , \lambda _ b \rightarrow b) classifierCart : isCartesian (Type \kappa , \lambda X \rightarrow X) classifier classifierCart _ = idIsEquiv
```

In other words, every polynomial functor p is classified by some polynomial universe. Moreover, if the classifying morphism  $p \hookrightarrow (Type \ \kappa \ , \ \lambda \ X \to X)$  is a Vertical embedding (i.e. a monomorphism in  $Poly^{Cart}$ ), then p itself is also a polynomial universe – for any pair of Cartesian morphisms  $f g : q \hookrightarrow p$ , since  $(Type \ \kappa \ , \ \lambda \ X \to X)$  is univalent, we have classifier  $\circ f \equiv classifier \circ g$ , but then since classifier is assumed to be a monomorphism, this implies that  $f \equiv g$ .

```
polyCartUniv : isVerticalEmbedding (Type \kappa , \lambda X \rightarrow X) classifier \rightarrow isUnivalent p polyCartUniv veclassifier cf cg = VertEmbedding \rightarrowPolyCartMono (Type \kappa , \lambda X \rightarrow X) classifierCart veclassifier cf cg (ua (compCartesian \_ cf classifierCart) (compCartesian \_ cg classifierCart))
```

#### open PolyCartUniv public

It follows that, for any type family  $P: Type \to Type$ , we can create a polynomial *sub-universe* of (Type ,  $\lambda X \to X$ ) by restricting to those types X for which there *merely* exists an inhabitant of P X.

```
module SubUniv \{\ell \mid \kappa\} (P : Type \ell \rightarrow Type \kappa) where
     subUniv : Poly (lsuc \ell) \ell
     subUniv = (\Sigma \text{ (Type } \ell) \ (\lambda \ X \rightarrow \| \ P \ X \|) \ , \ \lambda \ (X \ , \ \_) \rightarrow X)
     subUnivClassifierVertEmb :
           isVerticalEmbedding (Type \ell , \lambda X \rightarrow X) (classifier subUniv)
     subUnivClassifierVertEmb =
           Iso→isEquiv
                 ( (\lambda e → pairEq e \|-\|IsProp)
                 , ( ( \lambda e → (pairEq (ap (fst (classifier subUniv)) e)
                                            ||-||IsProp)
                               \equiv \langle ap (\lambda e' \rightarrow
                                             pairEq (ap (fst (classifier subUniv))
                                                            e) e')
                                         ||-||≡Contr > ( _
                                \equiv \langle (pairEq\eta e) \rangle
                                (e □)))
                      , (\lambda e → pairEq\beta1 e \|-\|IsProp) ) )
     subUnivIsUniv : isUnivalent subUniv
     subUnivIsUniv = polyCartUniv subUniv subUnivClassifierVertEmb
```

#### open SubUniv public

As a first example of a polynomial universe other than (Type ,  $\lambda$  X  $\rightarrow$  X), then, we may consider the polynomial universe of *propositions* P:

## module PropUniv where

```
P : Poly (lsuc lzero) lzero
P = subUniv isProp
```

If we write out explicitly the polynomial endofunctor defined by P we see that it has the following form:

$$y \mapsto \sum_{\phi: \mathbf{Prop}} y^{\phi}$$

This endofunctor (in fact it is a monad) is well-known in type theory by another name – the *partiality* monad. Specifically, this is the monad M whose kleisli morphisms  $A \rightarrow M$  B correspond to *partial functions* from A to B, that associate to each element A : A, a proposition def A indicating whether or not the value of A is defined at A, and a function val : def A and A be that takes a proof that A is defined at A to its value at A.

If we return to the original example of the polynomial universe (Type ,  $\lambda$  X  $\rightarrow$  X) we see that the associated polynomial endofunctor (which, by the above argument, is

also a monad) has a similar form.

$$y \mapsto \sum_{X: \mathbf{Type}} y^X$$

In this case, we can think of this as a "proof relevant" partiality monad M, such that a function  $f: A \to M$  B associates to each element a: A a *type* Def f a of proofs that f is defined at a, and a function val : Def f a  $\to$  B.6

More generally, we can say that, for any polynomial universe closed under dependent pair types, the associated monad will be a kind of (potentially proof-relevant) partiality monad, where the structure of the polynomial universe serves to dictate which types can count as *evidence* for whether or not a value is defined.

## **Rezk Completion**

In fact, we can show that for *any* polynomial functor, there exists a corresponding polynomial universe, using a familiar construct from the theory of categories in HoTT – the *Rezk Completion*. [AKS15] We will show that this construction allows us to quotient any polynomial functor to a corresponding univalent polynomial, i.e. a polynomial universe.

We obtain the Rezk completion of p as the image factorization in **Poly**<sup>Cart</sup> of the classifying morphism of p:

```
module RezkCompletion \{\ell \ \kappa\} (p : Poly \ell \ \kappa) where
```

The polynomial Rezk defined above can be seen to have the same form as a subuniverse of (Type,  $\lambda X \rightarrow X$ ); hence it is a polynomial universe, as desired.

```
RezkUniv : isUnivalent Rezk
RezkUniv = subUnivIsUniv (\lambda X \rightarrow \Sigma (fst p) (\lambda a \rightarrow (snd p a) \equiv X))
```

#### open RezkCompletion public

<sup>&</sup>lt;sup>6</sup>the conception of the monad determined by (Type ,  $\lambda$  X  $\rightarrow$  X) as a "proof relevant" partiality monad was communicated to the first author during private conversations with Jonathan Sterling.

As an example of how the Rezk completion allows us to "upgrade" a polynomial functor (a polynomial monad, even) into a polynomial universe, consider the following definition of the finite ordinals as a family of types indexed by the type Nat of natural numbers:

```
module FinUniv where
    open import Agda.Builtin.Nat
```

We first define the standard ordering on natural numbers:

```
data \_<\_: Nat \rightarrow Nat \rightarrow Type lzero where
zero< : \{n : Nat\} \rightarrow zero < suc n
succ< : \{n m : Nat\} \rightarrow n < m \rightarrow (suc n) < (suc m)
```

We then define the nth finite ordinal as the subtype of Nat consisting of all numbers m strictly less than n:

```
Fin : Nat \rightarrow Type lzero
Fin n = \Sigma Nat (\lambda m \rightarrow m < n)
```

From these data, we can straightforwardly define a polynomial as follows

```
\omega : Poly lzero lzero \omega = (Nat , Fin)
```

If we once again write out the polynomial endofunctor determined by these data

$$y \mapsto \sum_{n \in \mathbb{N}} y^{\{m \in \mathbb{N} \mid m < n\}}$$

we see that this functor has a familiar form – it is the *list monad* that maps a type y to the disjoint union of the types of n-tuples of elements of y, for all  $n \in \mathbb{N}$ .

As defined,  $\omega$  is not a polynomial universe; the type Nat is a set, and so for any n: Nat, the type n  $\equiv$  n is contractible, i.e. it has a single inhabitant, while the type of equivalences Fin n  $\simeq$  Fin n consists of all permutations of n elements, so these two types cannot be equivalent. However, we can now use the Rezk completion to obtain a polynomial universe from  $\omega$ .

```
Fin : Poly (lsuc lzero) lzero Fin = Rezk \omega
```

If we write out an explicit description of Fin, we see that it is the subuniverse of types X that are merely equivalent to some Fin n. In constructive mathematics, these types (they are necessarily sets) are known as *Bishop finite sets*. Hence the polynomial universe obtained by Rezk completion of the list monad is precisely the subuniverse of types spanned by (Bishop) finite sets.

## 5 Π-Types & Distributive Laws

We have so far considered how polynomial universes may be equipped with structure to interpret the unit type and dependent pair types. We have not yet, however, said much in the way of *dependent function types*. In order to rectify this omission, it will first be prudent to consider some additional structure on the category of polynomial functors – specifically a new functor  $\uparrow \uparrow [\_] : Tw(Poly) \times Poly \rightarrow Poly$  that plays a similar role for  $\Pi$  types as the composition  $\triangleleft : Poly \times Poly \rightarrow Poly$  played for  $\Sigma$  types, and which in turn bears a close connection to *distributive laws* in Poly.

#### 

The  $\uparrow \uparrow$  functor can be loosely defined as the solution to the following problem: given a polynomial universe u, find  $u \uparrow \uparrow u$  such that u classifies  $u \uparrow \uparrow u$  if and only if u has the structure to interpret  $\Pi$  types (in the same way that u classifies  $u \triangleleft u$  if and only if u has the structure to interpret  $\Sigma$  types). Generalizing this to arbitrary pairs of polynomials p = (A, B), q = (C, D) then yields the following formula for  $p \uparrow \uparrow q$ :

$$p \uparrow \uparrow q = \sum_{(a,f): \sum_{a:A} C^{B(a)}} y^{\prod_{b:B(a)} D(f(b))}$$

```
\begin{array}{l} - \uparrow \uparrow \_ : \ \forall \ \{\ell 0 \ \ell 1 \ \kappa 0 \ \kappa 1\} \rightarrow Poly \ \ell 0 \ \kappa 0 \rightarrow Poly \ \ell 1 \ \kappa 1 \\ \rightarrow Poly \ (\ell 0 \ \sqcup \ \kappa 0 \ \sqcup \ \ell 1) \ (\kappa 0 \ \sqcup \ \kappa 1) \end{array} (A \ , B) \ \uparrow \uparrow \ (C \ , D) = \\ ( \ \Sigma \ A \ (\lambda \ a \rightarrow B \ a \rightarrow C) \\ , \ (\lambda \ (a \ , \ f) \rightarrow (b \ : B \ a) \rightarrow D \ (f \ b))) \end{array}
```

```
\uparrow\uparrow Lens : \forall \{\ell 0 \ \ell 1 \ \ell 2 \ \ell 3 \ \kappa 0 \ \kappa 1 \ \kappa 2 \ \kappa 3\} 

\rightarrow \{p : Poly \ \ell 0 \ \kappa 0\} \ (r : Poly \ \ell 2 \ \kappa 2) 

\rightarrow \{q : Poly \ \ell 1 \ \kappa 1\} \ (s : Poly \ \ell 3 \ \kappa 3) 

\rightarrow (f : p \leftrightarrows r) \ (f' : r \leftrightarrows p) 

\rightarrow EqLens p (id p) (comp p f f') 

\rightarrow (g : q \leftrightarrows s) \rightarrow (p \uparrow\uparrow q) \leftrightarrows (r \uparrow\uparrow s)
```

$$\begin{array}{ccc} p & \rightarrow & q \\ = & = \\ p & \leftarrow & q \end{array}$$

<sup>&</sup>lt;sup>7</sup>To see why this is the right choice of morphism for which  $\uparrow\uparrow$  is functorial in its first argument, we note that pairs consisting of a morphism and a left inverse for it are equivalently the morphisms between identity morphisms in the *twisted arrow category* of **Poly**, i.e. diagrams of the following form:

By construction, the existence of a Cartesian lens  $(\pi, \pi\sharp): u \triangleleft u \leftrightarrows u$  effectively shows that u is closed under  $\Pi$ -types, since:

•  $\pi$  maps a pair (A , B) consisting of A : U and B :  $u(A) \to U$  to a term  $\pi(A,B)$  representing the corresponding  $\Pi$  type. This corresponds to the type formation rule

$$\frac{\Gamma \vdash A : \mathsf{Type} \qquad \Gamma, x : A \vdash B[x] \; \mathsf{Type}}{\Gamma \vdash \Pi x : A.B[x] \; \mathsf{Type}}$$

• The "elimination rule"  $\pi\sharp$  (A , B), for any pair (A , B) as above, maps an element  $f:\pi(A,B)$  to a function  $\pi\sharp$  (A , B)  $f:(a:u(A))\to u$  (B x) which takes an element x of A and yields an element of B x. This corresponds to the rule for function application:

$$\frac{\Gamma \vdash f : \Pi x : A.B[x] \qquad \Gamma \vdash a : A}{\Gamma \vdash f \ a : B[a]}$$

• Since  $\pi \sharp$  (A , B) is an equivalence, it follows that there is an inverse  $\pi \sharp^{-1}$  (A , B) : ((x : u(A))  $\rightarrow$  u(B(x))  $\rightarrow$  u( $\pi$ (A,B)), which corresponds to  $\lambda$ -abstraction:

$$\frac{\Gamma, x : A \vdash f[x] : B[x]}{\Gamma \vdash \lambda x . f[x] : \Pi x : A . B[x]}$$

• The fact that  $\pi \sharp^{-1}$  (A , B) is both a left and a right inverse to  $\pi \sharp$  then corresponds to the  $\beta$  and  $\eta$  laws for  $\Pi$  types.

$$(\lambda x. f[x]) a = f[a]$$
  $f = \lambda x. f x$ 

Although it is clear enough that the  $\uparrow\uparrow$  functor serves its intended purpose of characterizing  $\Pi$  types in polynomial universes, its construction seems somewhat more ad hoc than that of  $\triangleleft$ , which similarly characterized  $\Sigma$  types in polynomial universes while arising quite naturally from composition of polynomial functors. We would like to better understand what additional properties  $\uparrow\uparrow$  must satisfy, and how these in turn are reflected as properties of polynomial universes with  $\Pi$  types. In fact, we will ultimately show that this construction is intimately linked with a quite simple structure on polynomial universes u, namely a *distributive law* of u (viewed as a monad) over itself. Before that, however, we note some other key properties of  $\uparrow\uparrow$ .

Specifically, let  $\mathbf{Poly}_R$  be the category whose objects are polynomials and whose morphisms are lenses equipped with left inverses. Straightforwardly,  $\triangleleft$  restricts to a monoidal product on  $\mathbf{Poly}_R$ , since it is functorial in both arguments and must preserve

left/right inverses. Hence  $\uparrow\uparrow$  can be viewed as a functor  $\mathbf{Poly}_R \times \mathbf{Poly} \to \mathbf{Poly}$ . Then  $\uparrow\uparrow$  moreover naturally carries the structure of an *action* on  $\mathbf{Poly}$  of the monoidal category  $\mathbf{Poly}_R$  equipped with  $\triangleleft$ , in that there are natural transformations

$$y \uparrow \uparrow p \rightarrow p$$
 and  $(p \triangleleft q) \uparrow \uparrow r \rightarrow p \uparrow \uparrow (q \uparrow \uparrow r)$ 

which are moreover Cartesian:

```
module Unit\uparrow \uparrow \{\ell \ \kappa\} (p : Poly \ell \ \kappa) where
       y \uparrow \uparrow : (y \uparrow \uparrow p) \leq p
       y \uparrow \uparrow = ((\lambda (\_, a) \rightarrow a \text{ tt}), \lambda (\_, a) \text{ b tt } \rightarrow b)
       y \uparrow \uparrow Cart : isCartesian p y \uparrow \uparrow
       y \uparrow \uparrow Cart (_ , x) =
               Iso\rightarrowisEquiv ( (\lambda F \rightarrow F tt)
                                     , ( (\lambda a \rightarrow refl)
                                         \lambda b \rightarrow refl)
open Unit↑↑ public
module \triangleleft \uparrow \uparrow \{ \ell 0 \ \ell 1 \ \ell 2 \ \kappa 0 \ \kappa 1 \ \kappa 2 \} \ (p : Poly \ \ell 0 \ \kappa 0)
                   (q : Poly \ell 1 \kappa 1) (r : Poly \ell 2 \kappa 2) where
       \uparrow\uparrow Curry = ( (\lambda ((a, h), k))
                                 \rightarrow (a , (\lambda b \rightarrow ( (h b)
                                                           , (\lambda d \rightarrow k (b , d)))))
                        , (\lambda ((a, h), k) f (b, d) \rightarrow f b d))
       \uparrow\uparrowCurryCart : isCartesian (p \uparrow\uparrow (q \uparrow\uparrow r)) \uparrow\uparrowCurry
       \uparrow\uparrowCurryCart ((a , h) , k) =
               Iso\rightarrowisEquiv ( (\lambda f b d \rightarrow f (b , d))
                                     , ( (\lambda f \rightarrow refl)
```

open ⊲↑↑ public

The fact that \(\gamma\) Curry is Cartesian corresponds to the usual currying isomorphism that relating dependent functions types to dependent pair types:

,  $(\lambda \ f \rightarrow refl)$  )

$$\Pi(x,y): \Sigma x: A.B[x].C[x,y] \simeq \Pi x: A.\Pi y: B[x].C[x,y]$$

Similarly, ↑↑ is colax with respect to ▷ in its second argument, in that there are Cartesian natural transformations

$$p \uparrow \uparrow y \rightarrow y$$
 and  $p \uparrow \uparrow (q \triangleleft r) \rightarrow (p \uparrow \uparrow q) \triangleleft (p \uparrow \uparrow r)$ 

```
module \uparrow \uparrow Unit \{ \ell \ \kappa \} (p : Poly \ell \ \kappa) where
       \uparrow \uparrow y : (p \uparrow \uparrow y) \Leftrightarrow y
        \uparrow \uparrow yCart : isCartesian y \uparrow \uparrow y
        \uparrow \uparrow \gamma Cart (x, \gamma) =
               Iso\rightarrowisEquiv ( (\lambda x \rightarrow tt)
                                      , ( (\lambda a \rightarrow refl)
                                           , \lambda b → refl))
open ↑↑Unit public
module \uparrow \uparrow \triangleleft \{ \ell 0 \ \ell 1 \ \ell 2 \ \kappa 0 \ \kappa 1 \ \kappa 2 \} \ (p : Poly \ \ell 0 \ \kappa 0)
                    (q : Poly \ell 1 \kappa 1) (r : Poly \ell 2 \kappa 2) where
        \uparrow\uparrow \text{Distr} : (p \uparrow\uparrow (q \triangleleft r)) \leftrightarrows ((p \uparrow\uparrow q) \triangleleft (p \uparrow\uparrow r))
        \uparrow \uparrow Distr = ( (\lambda (a, h))
                                   \rightarrow ( (a , (\lambda b \rightarrow fst (h b)))
                                       , \lambda f \rightarrow (a , (\lambda b \rightarrow snd (h b) (f b))) ))
                         , (\lambda (a, h) (f, g) b \rightarrow (f b, g b))
        \uparrow\uparrowDistrCart : isCartesian ((p \uparrow\uparrow q) \triangleleft (p \uparrow\uparrow r)) \uparrow\uparrowDistr
        \uparrow\uparrowDistrCart (a , h) =
               Iso\rightarrowisEquiv ( (\lambda f \rightarrow ( (\lambda b \rightarrow fst (f b))
                                                        , (\lambda b \rightarrow snd (f b)))
                                       , ( (\lambda (f, g) \rightarrow refl)
                                           , (\lambda \ f \rightarrow refl) )
```

open ↑↑⊲ public

The fact that  $\uparrow\uparrow$ Distr is Cartesian corresponds to the distributive law of  $\Pi$  types over  $\Sigma$  types, i.e.

$$\Pi x : A.\Sigma y : B[x].C[x,y] \simeq \Sigma f : \Pi x : A.B[x].\Pi x : A.C[x,f(x)]$$

One may wonder, then, whether this distributive law is somehow related to a distributive law of g the monad structure on a polynomial universe u given by  $\Sigma$  types (as discussed in the previous section) over itself, i.e. a morphism

$$\mathfrak{u} \triangleleft \mathfrak{u} \subseteq \mathfrak{u} \triangleleft \mathfrak{u}$$

subject to certain laws. Indeed, given a Lens  $\pi:(u \uparrow \uparrow u) \hookrightarrow u$  (intuitively – corresponding to the structure of  $\Pi$  types in u), one can define a morphism of this form as follows:

```
distrLaw? : \forall \{\ell \ \kappa\} \ (u : Poly \ \ell \ \kappa) \rightarrow (u \ \uparrow \ u) \leftrightarrows u
\rightarrow (u \triangleleft u) \leftrightarrows (u \triangleleft u)
distrLaw? u \ (\pi \ , \ \pi \sharp) =
(\ (\lambda \ (a \ , b) \rightarrow \pi \ (a \ , b) \ , \ (\lambda \ x \rightarrow a))
, \ \lambda \ (a \ , b) \ (f \ , x) \rightarrow (x \ , \ (\pi \sharp \ ((a \ , b)) \ f \ x)) \ )
```

The question then becomes whether this morphism has the structure of a distributive law when u has the structure of a polynomial universe with  $\Sigma$  types, and  $\pi$  is Cartesian (i.e. u also has  $\Pi$  types). Answering this question in the affirmative shall be our task in the remainder of this section.

As a first step in this direction, we make a perhaps unexpected move of further generalizing the  $\uparrow\uparrow$  functor to a functor  $\mathsf{Tw}(\mathsf{Poly}) \times \mathsf{Poly} \to \mathsf{Poly}$ , where  $\mathsf{Tw}(\mathsf{Poly})$  is the *twisted arrow category* of  $\mathsf{Poly}$ , i.e. the category whose objects are lenses and whose morphisms are *twisted* commuting squares of the form

```
\begin{array}{ccc}
p & \to & p' \\
\downarrow & & \downarrow \\
q & \leftarrow & q'
\end{array}

\uparrow \uparrow [\_][\_] : \forall \{\ell \ \ell' \ \ell'' \ \kappa \ \kappa' \ \kappa''\}
                       \rightarrow (p : Poly \ell \kappa) (q : Poly \ell' \kappa')
                       \rightarrow (p \leftrightarrows q) \rightarrow (r : Poly \ell'' \kappa'')
                       \rightarrow Poly (\ell \sqcup \kappa \sqcup \ell'') (\kappa' \sqcup \kappa'')
(A, B) \uparrow \uparrow [(C, D)][(f, f \sharp)](E, F) =
      ((\Sigma A (\lambda a \rightarrow B a \rightarrow E))
      , (\lambda \ (a \ , \ \epsilon) \rightarrow (d : D \ (f \ a)) \rightarrow F \ (\epsilon \ (f\sharp \ a \ d))))
module \uparrow\uparrow[]Functor {\ell0 \ell1 \ell2 \ell3 \ell4 \ell5 \kappa0 \kappa1 \kappa2 \kappa3 \kappa4 \kappa5}
                   {p : Poly \ell 0 \kappa 0} {p' : Poly \ell 3 \kappa 3}
                   (q : Poly \ell 1 \kappa 1) \{q' : Poly \ell 4 \kappa 4\}
                   {r : Poly \ell 2 \kappa 2} (r' : Poly \ell 5 \kappa 5)
                   (f : p \leftrightarrows q) (f' : p' \leftrightarrows q')
                    (g : p \leftrightarrows p') (h : q' \leftrightarrows q) (k : r \leftrightarrows r')
                    (e : EqLens q f (comp q g (comp q f' h))) where
       \uparrow\uparrow[]Lens: (p \uparrow\uparrow[q][f]r) \leftrightarrows (p' \uparrow\uparrow[q'][f']r')
        ↑ [] Lens =
                ((\lambda (a, \gamma) \rightarrow (fst g a, \lambda x \rightarrow fst k (\gamma (snd g a x))))
                , \lambda (a , \gamma) F x \rightarrow
                       snd k (\gamma (snd f a x))
                             (transp (\lambda y \rightarrow snd r' (fst k (\gamma y)))
                                             (sym (snd e a x))
```

(F (snd h (fst f' (fst g a)))

(transp (snd q) (fst e a) x)))))

Straightforwardly, we have that  $p \uparrow \uparrow q = p \uparrow \uparrow [p][id p]q$ . In particular, we have  $\uparrow \uparrow Lens p p' q q' f f' e g = <math>\uparrow \uparrow [] Lens p p' p p' q q' (id p) (id p') f f' g e, which serves to motivate the definition of <math>\uparrow \uparrow Lens$  in terms of morphisms equipped with left inverses.

The functor  $\uparrow \uparrow [\_] [\_]$  defined above moreover preserves Cartesian morphisms in all of its arguments, and so restricts to a functor  $\mathsf{Tw}(\mathsf{Poly}^\mathsf{Cart}) \times \mathsf{Poly}^\mathsf{Cart} \to \mathsf{Poly}^\mathsf{Cart}$ .

```
↑ [] LensCart : isCartesian q h → isCartesian r' k
                   → isCartesian (p' \uparrow[ q' ][ f' ] r') \uparrow[]Lens
\uparrow\uparrow[]LensCart ch ck (a , \gamma) =
     compIsEquiv
           (PostCompEquiv (\lambda x \rightarrow snd k (\gamma (snd f a x)))
                                (\lambda x \rightarrow ck (\gamma (snd f a x))))
           (compIsEquiv
                (PostCompEquiv
                      (\lambda \times \rightarrow \text{transp } (\lambda \times \rightarrow \text{snd r' } (\text{fst k } (\gamma \times y)))
                                          (sym (snd e a x)))
                      (\lambda x \rightarrow transpIsEquiv (sym (snd e a x))))
                 (compIsEquiv
                      (PreCompEquiv (transp (snd q) (fst e a))
                                          (transpIsEquiv (fst e a)))
                      (PreCompEquiv (\lambda x \rightarrow snd h (fst f' (fst g a)) x)
                                          (ch (fst f' (fst g a)))))
```

#### open ↑ [] Functor public

Moreover, all the properties of  $\uparrow \uparrow$  noted above generalize to  $\downarrow \uparrow \uparrow \downarrow \uparrow$ . For instance, we now have natural transformations

```
y \uparrow \uparrow [y][id_y]p \rightarrow p and (p \triangleleft r) \uparrow \uparrow [q \triangleleft s][f \triangleleft g]t \rightarrow p \uparrow \uparrow [q][f](r \uparrow \uparrow [s][g]t)
```

as follows:

And similarly, we have natural transformations

As we shall now see, these structures on  $_{\uparrow}[_{]}[_{]}$  are intimately connected to a class of morphisms in **Poly**, which we call *distributors*.

## 5.2 Distributors

Given polynomials p,q,r,s, a *distributor* of p,q over r,s is a morphism of the form  $(p \triangleleft r) \leftrightarrows (s \triangleleft q)$  in **Poly**. The name "distributor" is here drawn from the fact that, given polynomial monads m,n with  $\eta_m:y\leftrightarrows m$ ,  $\eta_m:y\leftrightarrows n$  and  $\mu_m:(m\triangleleft m)\leftrightarrows m$ ,  $\mu_m:(n\triangleleft n)\leftrightarrows n$ , a *distributive law* of m over n consists of a distributor of n,n over n,n (i.e. a morphism  $\delta:(n\triangleleft m)\leftrightarrows m$ ) such that the following diagrams commute:

By inspection, it can be seen that all the composite morphisms required to commute by the above diagrams are themselves distributors of various forms. Understanding the closure properties of such distributors that give rise to these diagrams, then, will be a central aim of this section. **Remark 5.1.** Distributors as above arise in **EM**- $\mathbb{C}$ at<sup>#</sup> the Eilenberg-Moore completion [LS02] of the double category  $\mathbb{C}$ at<sup>#</sup> =  $\mathbb{C}$ omod(Poly) of polynomial comonads and bicomodules [SS24]. Indeed, there we find morphisms of the form  $m \triangleleft p \rightarrow p \triangleleft n$  for polynomial monads m, n, and distributive laws between monads are the formal monads in **EM**- $\mathbb{C}$ at<sup>#</sup>. Distributors also arise in what Lynch et al. call *effects handlers* [LSS23], where morphisms are of the form  $s \triangleleft c \rightarrow d \triangleleft s$  for polynomial comonads  $c, d \in \mathbb{C}$ at<sup>#</sup>.  $\diamond$ 

By function extensionality, we obtain the following type of equality proofs for distributors:

```
EqDistributor : \forall \{\ell 0 \ \ell 1 \ \ell 2 \ \ell 3 \ \kappa 0 \ \kappa 1 \ \kappa 2 \ \kappa 3\}
                             \rightarrow (p : Poly \ell0 \kappa0) (q : Poly \ell1 \kappa1)
                             \rightarrow (r : Poly \ell2 \kappa2) (s : Poly \ell3 \kappa3)
                             \neg (p \triangleleft r) \leftrightarrows (s \triangleleft q) \rightarrow (p \triangleleft r) \leftrightarrows (s \triangleleft q)
                             → Type (\ell0 \sqcup \ell1 \sqcup \ell2 \sqcup \ell3 \sqcup \kappa0 \sqcup \kappa1 \sqcup \kappa2 \sqcup \kappa3)
EqDistributor p q r s (f , f\sharp) (g , g\sharp) =
       (a : fst p) (\gamma : snd p a \rightarrow fst r)
       \rightarrow \Sigma (fst (f (a , \gamma)) \equiv fst (g (a , \gamma)))
               (\lambda \text{ el} \rightarrow (x : \text{snd s (fst (f (a , <math>\gamma)))}))
                             \rightarrow \Sigma ((snd (f (a , \gamma)) x)
                                     \equiv (snd (g (a , \gamma))
                                                  (transp (snd s) e1 x)))
                                     (\lambda \ e2 \rightarrow (y : snd \ q \ (snd \ (f \ (a \ , \ \gamma)) \ x))
                                                    \rightarrow (f# (a , \gamma) (x , y))
                                                       \equiv (g# (a , \gamma)
                                                                   ( (transp (snd s) e1 x)
                                                                   , (transp (snd q) e2 y)))))
```

Moreover, for any polynomial u with  $\pi$ : ( $u \uparrow \uparrow u$ )  $\leftrightarrows u$ , the morphism distrLaw? u  $\pi$  defined above is a distributor of u, u over itself. In fact, we can straightforwardly generalize the construction of distrLaw? to a transformation

$$(p \uparrow \uparrow [q][f]r) \leftrightharpoons s \implies (p \triangleleft r) \leftrightharpoons (s \triangleleft q)$$

as follows:

Hence to show that the above-given diagrams commute for the candidate distributive law distrLaw? u  $\pi$  given above, it suffices to show that the distributors required to commute by these diagrams themselves arise – under the above-defined transformation – from Cartesian morphisms of the form p  $\uparrow [ q ] [ f ] r \hookrightarrow u$ , which, if u is a polynomial universe, are necessarily equal.

First of all, any distributor  $(p \triangleleft r) \leftrightarrows (s \triangleleft q)$  may be extended along morphisms  $p' \leftrightarrows p$ ,  $q \leftrightarrows q'$ ,  $r' \leftrightarrows r$ ,  $s \leftrightarrows s'$  to a distributor  $(p' \triangleleft r') \leftrightarrows (s' \triangleleft q')$  by forming the composite

 $p' \triangleleft r' \rightarrow p \triangleleft r \rightarrow s \triangleleft q \rightarrow s' \triangleleft q'$ 

```
module DistributorLens {\ell 0 \ \ell 1 \ \ell 2 \ \ell 3 \ \ell 4 \ \ell 5 \ \ell 6 \ \ell 7}
\kappa 0 \ \kappa 1 \ \kappa 2 \ \kappa 3 \ \kappa 4 \ \kappa 5 \ \kappa 6 \ \kappa 7}
{p : Poly \ \ell 0 \ \kappa 0} {p' : Poly \ \ell 4 \ \kappa 4}
{q : Poly \ \ell 1 \ \kappa 1} (q' : Poly \ \ell 5 \ \kappa 5)
(r : Poly \ \ell 2 \ \kappa 2) {r' : Poly \ \ell 6 \ \kappa 6}
{s : Poly \ \ell 3 \ \kappa 3} (s' : Poly \ \ell 7 \ \kappa 7)
(g : p' \hookrightarrow p) (h : q \hookrightarrow q')
(g : p' \hookrightarrow p) (h : q \hookrightarrow q')
(g : r' \hookrightarrow r) (g : p' \hookrightarrow p) (g : q' \circ q')
distrLens : (g : q' \circ q') (g : q' \circ q' \circ q')
```

 $(comp ((s' \triangleleft q')) j$ 

 $(1 \triangleleft [q']h)$ 

The corresponding construction on morphisms out of  $\uparrow \uparrow [\_] [\_]$  is given by forming the composite

 $p' \uparrow \uparrow [q'] [h \circ f \circ q] r' \hookrightarrow p \uparrow \uparrow [q] [f] r \hookrightarrow s \hookrightarrow s'$ 

#### open DistributorLens public

Similarly, there are two distinct ways of composing distributors:

1. Given distributors  $p \triangleleft s \leftrightarrows t \triangleleft q$  and  $q \triangleleft u \leftrightarrows v \triangleleft r$ , we obtain a distributor  $p \triangleleft (s \triangleleft u) \leftrightarrows (t \triangleleft v) \triangleleft r$  as the composite

$$p \triangleleft (s \triangleleft u) \simeq (p \triangleleft s) \triangleleft u \longrightarrow (t \triangleleft q) \triangleleft u \simeq t \triangleleft (q \triangleleft u) \longrightarrow t \triangleleft (v \triangleleft r) \simeq (t \triangleleft v) \triangleleft r$$

$$\text{module DistributorComp1} \ \{ \ell 0 \ \ell 1 \ \ell 2 \ \ell 3 \ \ell 4 \ \ell 5 \ \ell 6 \ \kappa 0 \ \kappa 1 \ \kappa 2 \ \kappa 3 \ \kappa 4 \ \kappa 5 \ \kappa 6 \}$$

$$\{ p : \text{Poly } \ell 0 \ \kappa 0 \} \ \{ q : \text{Poly } \ell 1 \ \kappa 1 \} \ (r : \text{Poly } \ell 2 \ \kappa 2 )$$

{s : Poly  $\ell$ 3  $\kappa$ 3} {t : Poly  $\ell$ 4  $\kappa$ 4} (u : Poly  $\ell$ 5  $\kappa$ 5) {v : Poly  $\ell$ 6  $\kappa$ 6} where

 $\begin{array}{l} \mbox{distrComp1} : (p \lessdot s) \leftrightarrows (t \lessdot q) \rightarrow (q \lessdot u) \leftrightarrows (v \lessdot r) \\ & \rightarrow (p \lessdot (s \lessdot u)) \leftrightarrows ((t \lessdot v) \lessdot r) \\ \mbox{distrComp1} \ h \ k = \\ \mbox{comp} \ ((t \lessdot v) \lessdot r) \ ( \lnot assoc^{-1} \ p \ s \ u) \\ \mbox{(comp} \ ((t \lessdot v) \lessdot r) \ (h \lessdot [\ u \ ] \ (id \ u)) \\ \mbox{(comp} \ ((t \lessdot v) \lessdot r) \ ( \lnot assoc \ t \ q \ u) \\ \mbox{(comp} \ ((t \lessdot v) \lessdot r) \ ((id \ t) \lessdot [\ (v \lessdot r) \ ] \ k) \end{array}$ 

The corresponding construction on morphisms (p  $\uparrow [ q ] [ f ] s$ )  $\hookrightarrow$  t and (q  $\uparrow [ r ] [ g ] u$ )  $\hookrightarrow$  v is to form the following composite with the colaxator of  $\_\uparrow [\_] [\_] \_:$ 

 $(\triangleleft assoc^{-1} t v r)))$ 

$$p \! \uparrow \uparrow \! [r][g \circ f](s \triangleleft u) \leftrightarrows (p \! \uparrow \uparrow \! [q][f]s) \triangleleft (q \! \uparrow \uparrow \! [r][g]u) \leftrightarrows t \triangleleft v$$

↑↑→DistributorComp1 :

↑↑→DistributorComp1≡ :

#### open DistributorComp1 public

2. Given distributors  $p \triangleleft u \leftrightharpoons v \triangleleft q$  and  $r \triangleleft t \leftrightharpoons u \triangleleft s$ , we obtain a distributor  $(p \triangleleft r) \triangleleft t \leftrightharpoons v \triangleleft (q \triangleleft s)$  as the composite

$$(p \triangleleft r) \triangleleft t \simeq p \triangleleft (r \triangleleft t) \leftrightarrows p \triangleleft (u \triangleleft s) \simeq (p \triangleleft u) \triangleleft s \leftrightarrows (v \triangleleft q) \triangleleft s \simeq v \triangleleft (q \triangleleft s)$$

```
module DistributorComp2
```

```
 \{ \emptyset \ \ell 1 \ \ell 2 \ \ell 3 \ \ell 4 \ \ell 5 \ \ell 6 \ \kappa 0 \ \kappa 1 \ \kappa 2 \ \kappa 3 \ \kappa 4 \ \kappa 5 \ \kappa 6 \} 
 \{ p : Poly \ \ell 0 \ \kappa 0 \} \ \{ q : Poly \ \ell 1 \ \kappa 1 \} 
 \{ r : Poly \ \ell 2 \ \kappa 2 \} \ (s : Poly \ \ell 3 \ \kappa 3) 
 (t : Poly \ \ell 4 \ \kappa 4) \ \{ u : Poly \ \ell 5 \ \kappa 5 \} 
 \{ v : Poly \ \ell 6 \ \kappa 6 \} \ \textbf{where} 
 \text{distrComp2} : (r \triangleleft t) \leftrightarrows (u \triangleleft s) \rightarrow (p \triangleleft u) \leftrightarrows (v \triangleleft q) 
 \rightarrow ((p \triangleleft r) \triangleleft t) \leftrightarrows (v \triangleleft (q \triangleleft s)) 
 \text{distrComp2 h } k = 
 \text{comp} \ (v \triangleleft (q \triangleleft s)) \ (\exists assoc \ p \ r \ t) 
 (\text{comp} \ (v \triangleleft (q \triangleleft s)) \ (\exists assoc^{-1} \ p \ u \ s) 
 (\text{comp} \ (v \triangleleft (q \triangleleft s)) \ (\exists assoc^{-1} \ p \ u \ s) 
 (\exists assoc \ v \ q \ s)))
```

The corresponding construction on morphisms ( $p \uparrow [q][f]u$ )  $\Rightarrow v$  and ( $r \uparrow [s][g]t$ )  $\Rightarrow u$  is to form the following composite with the morphism  $\uparrow []$ Curry defined above:

```
(p \triangleleft r) \uparrow \uparrow [q \triangleleft s] [f \triangleleft q] t \Leftrightarrow p \uparrow \uparrow [q] [f] (r \uparrow \uparrow [s] [q] t) \Leftrightarrow p \uparrow \uparrow [q] [f] u \Leftrightarrow v
↑↑→DistributorComp2 :
        \{f : p \subseteq q\} \{g : r \subseteq s\}
        \rightarrow (r \uparrow [s][g]t) <math>\rightleftharpoons u
        \rightarrow (p \uparrow \uparrow [q][f]u) <math>\rightleftharpoons v
        \rightarrow ((p \triangleleft r) \uparrow \uparrow [ (q \triangleleft s) ][ f \triangleleft \triangleleft [ s ] g ] t) \leftrightarrows v
\uparrow\uparrow \rightarrow DistributorComp2 \{f = f\} \{g = g\} h k =
        comp v (\uparrow\uparrow[]Curry p q r s t f g)
                  (comp \ v \ (\uparrow f)]Lens q u f f
                                                     (id p) (id q) h
                                                      ((\lambda a \rightarrow refl)
                                                      , (\lambda \text{ a d} \rightarrow \text{refl}))
                              k)
↑↑→DistributorComp2≡ :
        \{f : p \leq q\} \{g : r \leq s\}
        \rightarrow (h : (r \uparrow [s][g]t) \Leftrightarrow u)
        \rightarrow (k : (p \uparrow [q][f]u) \leftrightarrows v)
        → (distrComp2 (↑↑→Distributor s t h)
                                     (\uparrow\uparrow \rightarrow Distributor q u k))
            \equiv \uparrow \uparrow \rightarrow Distributor (q \triangleleft s) t
                                             (↑↑→DistributorComp2 h k)
\uparrow \uparrow \rightarrow DistributorComp2 \equiv h k = refl
```

open DistributorComp2 public

Likewise, there are two corresponding notions of "identity distributor" on a polynomial p, the first of which is given by the following composition of unitors for <:

$$p \triangleleft y \simeq p \simeq y \triangleleft p$$

and the second of which is given by the inverse such composition

```
y \triangleleft p \simeq p \simeq p \triangleleft y
```

module DistributorId  $\{\ell \ \kappa\}$  (p : Poly  $\ell \ \kappa$ ) where

```
distrId1 : (p \triangleleft y) \leftrightarrows (y \triangleleft p)
distrId1 = comp (y \triangleleft p) (\triangleleftunitr p) (\triangleleftunitl^{-1} p)
distrId2 : (y \triangleleft p) \leftrightarrows (p \triangleleft y)
distrId2 = comp (p \triangleleft y) (\triangleleftunitl p) (\triangleleftunitr^{-1} p)
```

The corresponding morphisms  $p \uparrow [p][id p] y \hookrightarrow y$  and  $y \uparrow [y][id y]$   $p \hookrightarrow p$  are precisely the maps  $\uparrow []y$  and  $y \uparrow []$  defined above, respectively:

```
\uparrow \rightarrow DistributorId1 \equiv : distrId1 \equiv \uparrow \uparrow \rightarrow Distributor p y (\uparrow \uparrow [] y p p (id p))
\uparrow \rightarrow DistributorId1 \equiv = ref1
\uparrow \rightarrow DistributorId2 \equiv : distrId2 \equiv \uparrow \uparrow \rightarrow Distributor y p (y \uparrow \uparrow [] p)
\uparrow \rightarrow DistributorId2 \equiv = ref1
```

### open DistributorId public

It can thus be seen that the above operations defined on distributors are preicsely those occurring in the diagrams for a distributive law given above, and moreover, these all have corresponding constructions on morphisms out of  $_{\uparrow}$ [\_][\_]\_\_, all of which preserve Cartesian morphisms. Hence if  $\pi: u \uparrow u \downarrow u$  is Cartesian, all of the morphisms involving  $_{\uparrow}$ [\_][\_]\_ corresponding to those required to commute in order for distrLaw?  $u \pi$  to be a distributive law will be Cartesian, and so if u is a polynomial universe, these will all automatically be equal to one another.

```
ap\uparrow\uparrow-Distributor : \forall {\ell0 \ell1 \ell2 \ell3 \kappa0 \kappa1 \kappa2 \kappa3}
 \rightarrow (p : Poly \ \ell 0 \ \kappa 0) \ (q : Poly \ \ell 1 \ \kappa 1)
 \rightarrow (r : Poly \ \ell 2 \ \kappa 2) \ (s : Poly \ \ell 3 \ \kappa 3) \ (f : p \leftrightarrows q)
 \rightarrow (h \ k : (p \uparrow \uparrow [ \ q \ ] [ \ f \ ] \ r) \leftrightarrows s) \rightarrow EqLens \ s \ h \ k
 \rightarrow EqDistributor \ p \ q \ r \ s \ (\uparrow \uparrow \rightarrow Distributor \ q \ r \ h)
 (\uparrow \uparrow \rightarrow Distributor \ q \ r \ k)
 ap \uparrow \uparrow \rightarrow Distributor \ p \ q \ r \ s \ f \ h \ k \ (e \ , e \sharp) \ a \ \gamma =
 (e \ (a \ , \gamma)
 , \lambda \ x \rightarrow (refl \ , (\lambda \ y \rightarrow pairEq \ refl \ (coAp \ (e \sharp \ (a \ , \gamma) \ x) \ y)) \ )
```

```
module DistrLaw \{\ell \ \kappa\} (u : Poly \ \ell \ \kappa) (univ : isUnivalent u)
                         (\eta : y \leftrightarrows u) (c\eta : isCartesian u \eta)
                         (\sigma : (u \triangleleft u) \subseteq u) (c\sigma : isCartesian u \sigma)
                         (\pi : (u \uparrow \uparrow u) \leftrightarrows u) (c\pi : isCartesian u \pi) where
      distrLaw1 : EqDistributor u u (u \triangleleft u) u
                               (distrLens u (u \triangleleft u) u (id u) (id u) (id (u \triangleleft u)) \sigma
                                                (distrComp1 u u (distrLaw? u \pi)
                                                                         (distrLaw? u \pi)))
                               (distrLens u u u (id u) (id u) \sigma (id u)
                                                (distrLaw? u \pi))
      distrLaw1 =
            ap\uparrow\uparrow→Distributor u u (u \triangleleft u) u (id u)
                   (comp u (comp (u \triangleleft u) (\uparrow Distr u u u) (\pi \triangleleft [u] \pi)) \sigma)
                   (comp u (\uparrow []Lens u u (id u) (id u) (id u) (id u) \sigma
                                             ((\lambda \ a \rightarrow refl), (\lambda \ a \ d \rightarrow refl)))
                            \pi)
                   (univ (compCartesian u
                                     (compCartesian (u \triangleleft u)
                                           (↑DistrCart u u u)
                                           (\triangleleft Cart u u c\pi c\pi))
                                     c\sigma)
                            (compCartesian u)
                               \left( \prod [] \text{LensCart } u \ u \ (\text{id } u) \ (\text{id } u) \ (\text{id } u) \ (\text{id } u) \ \sigma \right)
                                     ((\lambda \ a \rightarrow refl), (\lambda \ a \ d \rightarrow refl))
                                     (idCart u) c\sigma)
                               c\pi)
      distrLaw2 : EqDistributor (u \triangleleft u) u u u
                               (distrLens u u u (id (u \triangleleft u)) \sigma (id u) (id u)
                                                (distrComp2 u u (distrLaw? u \pi)
                                                                         (distrLaw? u \pi)))
                               (distrLens u u \sigma (id u) (id u) (id u)
                                                (distrLaw? u \pi))
      distrLaw2 =
            ap\uparrow\uparrow→Distributor (u \triangleleft u) u u u \sigma
                  (comp u)
                         (comp (u \uparrow \uparrow u))
                               (comp (u \uparrow (u \uparrow u))
                                     (\uparrow \uparrow [] Lens u u \sigma (id (u \triangleleft u))
                                           (id (u \triangleleft u)) \sigma (id u)
                                           ((\lambda \ a \rightarrow refl), (\lambda \ a \ d \rightarrow refl)))
                                     (\uparrow \uparrow Curry u u u))
                               (\uparrow\uparrowLens u u (id u) (id u)
```

```
((\lambda \ a \rightarrow refl), (\lambda \ a \ d \rightarrow refl))
                                  \pi))
                  \pi)
            (comp u (\uparrow\uparrow[]Lens u u \sigma (id u) \sigma (id u) (id u)
                                     ((\lambda \ a \rightarrow refl), (\lambda \ a \ d \rightarrow refl)))
                        \pi)
            (univ (compCartesian u
                        (compCartesian (u \uparrow \uparrow u)
                              (compCartesian (u \uparrow (u \uparrow u))
                                    (\uparrow \uparrow [] LensCart u u \sigma (id (u \triangleleft u))
                                           (id (u \triangleleft u)) \sigma (id u)
                                          ((\lambda \ a \rightarrow refl), (\lambda \ a \ d \rightarrow refl))
                                          c\sigma (idCart u))
                                    (\uparrow\uparrow CurryCart u u u))
                              \left( \uparrow \right) []LensCart u u (id u) (id u) (id u) (id u) \pi
                                                  ((\lambda \ a \rightarrow refl), (\lambda \ a \ d \rightarrow refl))
                                                  (idCart u) c\pi))
                        c\pi)
                     (compCartesian u)
                        (\uparrow [] LensCart u u \sigma (id u) \sigma (id u) (id u)
                              ((\lambda \ a \rightarrow refl), (\lambda \ a \ d \rightarrow refl))
                              (idCart u) (idCart u))
                        c\pi)
distrLaw3 : EqDistributor u u y u
                        (distrLens u y u (id u) (id u) (id y) \eta
                                         (distrId1 u))
                        (distrLens u u u (id u) (id u) \eta (id u)
                                         (distrLaw? u \pi))
distrLaw3 =
      ap\uparrow\uparrow→Distributor u u y u (id u)
            (comp \ u \ (\uparrow \uparrow y \ u) \ \eta)
            (comp u (\uparrow \uparrowLens u u (id u) (id u)
                                   ((\lambda \ a \rightarrow refl), (\lambda \ a \ d \rightarrow refl)) \eta) \pi)
            (univ (compCartesian u (\uparrow \uparrow yCart u) c\eta)
                     (compCartesian u)
                        (\uparrow [] LensCart u u (id u) (id u) (id u) (id u) \eta
                                            ((\lambda \ a \rightarrow refl), (\lambda \ a \ d \rightarrow refl))
                                            (idCart u) c\eta)
                        c\pi)
distrLaw4 : EqDistributor y u u u
                        (distrLens u u (id y) \eta (id u) (id u)
                                         (distrId2 u))
```

```
(distrLens u u \eta (id u) (id u) (id u)
                                          (distrLaw? u \pi))
distrLaw4 =
      ap\uparrow\uparrow→Distributor y u u u \eta
            (comp u (\uparrow\uparrow[]Lens u u \eta (id v) (id v) \eta (id u)
                                       ((\lambda \ a \rightarrow refl), (\lambda \ a \ d \rightarrow refl)))
                         (y \uparrow \uparrow u)
            (comp u (\uparrow\uparrow[]Lens u u \eta (id u) \eta (id u) (id u)
                                       ((\lambda \ a \rightarrow refl), (\lambda \ a \ d \rightarrow refl)))
                         \pi)
            (univ (compCartesian u
                         (\uparrow\uparrow[]LensCart u u \eta (id y) (id y) \eta (id u)
                                             ((\lambda \ a \rightarrow refl), (\lambda \ a \ d \rightarrow refl))
                                             c\eta (idCart u))
                         (y \uparrow \uparrow Cart u))
                      (compCartesian u)
                         (\uparrow [] LensCart u u \eta (id u) \eta (id u) (id u)
                                             ((\lambda \ a \rightarrow refl), (\lambda \ a \ d \rightarrow refl))
                                             (idCart u) (idCart u))
                         c\pi)
```

Hence distrLaw? u  $\pi$  is a distributive law, as desired (and moreover, all of the higher coherences of an  $\infty$ -distributive law could be demonstrated, following this same method.)

# 6 Further Structures on Polynomial Universes

In closing, we turn to briefly consider whether and how some additional type-theoretic constructs may be defined for natural models / polynomial universes in the language of polynomial functors, starting with the concept of a universe itself.

## 6.1 The Shift Operator & Universes

Throughout this paper, we have made extensive use of universes of types. A natural question to ask, then, is when the type theory presented by a polynomial universe itself contains another such universe as a type within itself.

For this purpose, let v, u be polynomial universes with v = (V, E1V) and u = (U, E1U). If there is a (necessarily unique) Cartesian morphism  $v \subseteq u$ , then it follows that every type family classified by v is also classified by u, by composition of Cartesian morphisms. However, what we want in this case is the stronger property that v is somehow represented as a type within u.

For this purpose, we define the following *shift* operation that takes a polynomial p

= (A , B) to the polynomial shift p = (
$$\top$$
 ,  $\lambda$  \_  $\rightarrow$  A). Equivalently, this is  $\begin{bmatrix} p \\ y \end{bmatrix}$ ,

where 
$$\begin{bmatrix} - \\ - \end{bmatrix}$$
 is the left coclosure of  $\triangleleft$  defined in [NS24].  
shift :  $\forall \{\ell \ \kappa\} \rightarrow \text{Poly} \ \ell \ \kappa \rightarrow \text{Poly lzero} \ \ell$   
shift (A , \_) = ( $\top$  ,  $\lambda$  \_  $\rightarrow$  A)

By construction, then, if there is a Cartesian morphism  $(v, v\sharp)$ : shift  $(V, ElV) \hookrightarrow (U, ElU)$ , it follows that:

• There is a type v tt : *U*; type theoretically, this corresponds to a type formation rule of the form

$$\Gamma \vdash \mathcal{V}$$
 Type

We think of *V* as a type whose elements are "codes" for other types.

• There is a function  $v \sharp tt : ElU (v tt) \rightarrow V$ , corresponding to the rule

$$\frac{\Gamma \vdash e : V}{\Gamma \vdash [e] \text{ Type}}$$

which decodes a code contained in *V* to its corresponding type.

• There is a function  $v^{+1}$  tt :  $V \to ElU$  (v tt), corresponding to the rule

$$\frac{\Gamma \vdash A \text{ Type} \qquad A \text{ is classifed by } \mathfrak{v}}{\Gamma \vdash \lfloor A \rfloor : \mathcal{V}}$$

that assigns a code to each type classified by v (note that this restriction to types classified by v is necessary to avoid the paradoxes that would arise from having a type universe that contained itself.)

Such that the following equations hold

$$\lceil \lfloor A \rfloor \rceil = A$$
  $e = \lfloor \lceil e \rceil \rfloor$ 

# 6.2 The (-)<sup>=</sup> Operator & Extensional Identity Types

Another key construct of dependent type theory which has figured prominently in the foregoing development of polynomial universes, but which we have not yet shown how to internalize in such universes, is the construction of *identity types*. To some extent, this choice has been deliberate, as the theory of identity types is arguably one of the most complex aspects of dependent type theory, as evidenced by the fact that research into this topic ultimately precipitated the development of homotopy type theory. For this very reason, however, an account of the semantics of dependent type theory without at least some indication of its application to the theory of identity types would be incomplete.

Readers familiar with dependent type theory may be aware that an initial complication posed by the theory of identity types is that these types come in two flavors: extensional and intensional. Extensional identity types reflect propositional equality (i.e. the existence of an inhabitant for the type  $a \equiv b$ ) into judgmental equality

(i.e. the metatheoretic proposition that a = b) and additionally regard all such proofs of identity as themselves identical. It follows that these identity types carry none of the homotopical information afforded by the alternative – intensional identity types, which are the sort which we have so far used in this paper. However, when working within such a homotopical framework, wherein metatheoretic equality need not be a mere proposition, there exists the possibility of defining extensional identity types in a polynomial universe so as to enable the aforementioned reflection while still allowing proofs of identity to carry higher-dimensional data.

For this purpose, let u=(U), E1) be a polynomial universe. We wish to establish under what conditions u would be closed under the formation of "identity types" for the types classified by it. Solving this problem in essentially the same manner as led to the definiiton of the  $\uparrow\uparrow$  functor in the previous section yields the following construction, that maps p=(A,B) to the polynomial  $p^=(\Sigma,A)$  ( $\lambda,B$ )  $\lambda,B$  ( $\lambda,B$ 

\_= : 
$$\forall$$
 {ℓ κ} → Poly ℓ κ → Poly (ℓ ⊔ κ) κ  
(A , B)= = (Σ A ( $\lambda$  a → B a × B a) ,  $\lambda$  (\_ , (b1 , b2)) → b1 ≡ b2)

If there is a Cartesian morphism ( $\epsilon$ ,  $\epsilon \sharp$ ):  $u^{=} \leftrightarrows u$  then:

• For each type A: U with elements  $a_0$ ,  $a_1: El A$ , there is a type  $Eq(A, a_0, a_1): U$ . Type theoretically, this corresponds to the type formation rule

$$\frac{\Gamma \vdash A \text{ Type} \qquad \Gamma \vdash a_0 : A \qquad \Gamma \vdash a_1 : A}{\Gamma \vdash a_0 \equiv_A a_1 \text{ Type}}$$

• For each A: U and  $a_0$ ,  $a_1$ : El A as above, there is a function El Eq(A,  $a_0$ ,  $a_1$ )  $\rightarrow a_0 \equiv a_1$  corresponding to the *reflection rule* 

$$\frac{\Gamma \vdash e : a_0 \equiv_A a_1}{\Gamma \vdash a_0 = a_1}$$

that converts an inhabitant of the propositional equality into a proof of the corresponding judgmental equality.

• Likewise, there is a function  $a_0 \equiv a_1 \rightarrow El Eq(A, a_0, a_1)$  corresponding to the reflexivity rule

$$\frac{\Gamma \vdash a_0 = a_1}{\Gamma \vdash \mathsf{refl} : a_0 \equiv_A a_1}$$

that produces an inhabitant of the propositional equality given a proof of the corresponding judgmental equality.

• Such that the above two functions/rules are mutually inverse.

What is missing from the above description of extensional identity types is the following rule

$$\frac{\Gamma \vdash e_1 : a_0 \equiv_A a_1}{\Gamma \vdash e_1 = e_2} \stackrel{\Gamma \vdash e_2 : a_0 \equiv_A a_1}{= e_2}$$

which says that all inhabitants of the identity type are themselves identical (i.e. the identity type is a *mere proposition*). This rule would be validated if we additionally required u to have the property that, for all types A: U, the type ELA is a set (hence for any ab: ELA, the type ab: BLA is a mere proposition.) However, if we do not make this requirement, this opens the possibility of having a model of extensional type theory, modulo the above rule, wherein proofs of equality still carry homotopical information – a potentially new direction in research on the semantics of identity types.

### 6.2.1 A Note on Intensional Identity Types & Inductive Types

Attempting to account for *intensional* rather than *extensional* identity types in the language of polynomial functors is rather more complicated, however. As mentioned in Section 2, the inhabitants of intensional identity types are inductively generated from the constructor ref1, corresponding to reflexivity. The problem with such inductive generation of data from constructors – from the point of view taken in this paper – is that it characterizes types in terms of their introduction forms, rather than their elimination forms. In type theoretic jargon, we say that intensional identity types, and inductive types more generally are *positive*, whereas all of the types we have considered so far are *negative*, in that they are characterized by their elimination forms. The universal properties of such negative types are therefore *mapping-in* properties, which are naturally described in terms of presheaves, which we have taken as our intended model for the development of this paper. By contrast, however, the universal properties of positive types are given by *mapping-out* properties, which are rather described in terms of (the opposite category of) *co-presheaves*.

As an illustrative example, let us consider the rather simpler case of (binary) coproducts, which are naturally regarded as positive types characterized by the left and right injections  $A \rightarrow A + B$  and  $B \rightarrow A + B$ . one might think to define binary coproducts on a polynomial universe u in the following way:

The *product* of two polynomial functors  $p = \sum_{a:A} y^{B[a]}$  and  $q = \sum_{c:C} y^{D[c]}$  can be calculated as follows:

$$\left( \sum_{a:A} y^{B[a]} \right) \times \left( \sum_{c:C} y^{D[c]} \right)$$

$$\simeq \sum_{(a,c):A \times C} y^{B[a]} \times y^{D[c]}$$

$$\simeq \sum_{(a,c):A \times C} y^{B[a]+D[c]}$$

Hence one might think to define binary coproducts on a polynomial universe u = (U , E1) by asking there to be a Cartesian morphism  $u \times u \subseteq u$ , since this would mean that for every pair of types (A , B) :  $U \times U$ , there is a type plus(A , B) :  $U \times U$  such that  $E1(plus(A , B)) \simeq E1 A + E1 B$ .

However, from the perspective of natural models, this condition is too strong. Given a category of contexts  $\mathcal{C}$ , the category  $\mathbf{Set}^{\mathcal{C}^{op}}$  of presheaves on  $\mathcal{C}$  is the free cocompletion of  $\mathcal{C}$ , which means that requiring  $\mathcal{C}$  to be closed under taking binary coproducts of representables in  $\mathbf{Set}^{\mathcal{C}^{op}}$  means not only that  $\mathcal{C}$  has all binary coproducts, but that in fact all such coproducts in  $\mathcal{C}$  are *free*.

Hence it remains to be seen if there can be found a general way of correctly expressing such "positive" type-theoretic concepts as for polynomial universes and natural models

in the language of polynomial functors. We hope to continue investigations into these and related questions in future work.

### 6.3 Conclusion

In this paper, we have advanced a simplified and unified account of the categorical semantics of dependent type theory by expressing the core concepts of natural models entirely within the framework of polynomial functors in HoTT. By utilizing HoTT, we have been able strike an ideal balance between issues of strictness and higher-dimensional coherence that have bedeviled previous accounts. This shift not only streamlines the presentation of the semantics of dependent type theory, but also reveals additional structures thereof, such as the self-distributive law governing the interaction between dependent products and sums.

However, there remain many open questions regarding the further development of this framework, particularly with respect to *positive* type-theoretic constructs such as coproducts, inductive types, and intensional identity types. Further work is needed to explore whether polynomial functors can provide a fully general account of these concepts. We look forward to continuing these investigations, with the aim of extending the unification presented here to encompass a wider range of type-theoretic phenomena.

### References

- [AKS15] Benedikt Ahrens, Krzysztof Kapulkin, and Michael Shulman. "Univalent Categories and the Rezk Completion". In: *Extended Abstracts Fall 2013*. Ed. by Maria del Mar González, Paul C. Yang, Nicola Gambino, and Joachim Kock. Cham: Springer International Publishing, 2015, pp. 75–76 (cit. on p. 27).
- [AN18] Steve Awodey and Clive Newstead. "Polynomial pseudomonads and dependent type theory". In: *arXiv* (2018). eprint: 1802.00997 (cit. on pp. 2, 20).
- [Awo14] Steve Awodey. "Natural models of homotopy type theory". In: (2014). eprint: arXiv:1406.3219 (cit. on pp. 2, 5, 20).
- [LS02] Stephen Lack and Ross Street. "The formal theory of monads II". In: *Journal of pure and applied algebra* 175.1-3 (2002), pp. 243–265 (cit. on p. 36).
- [LSS23] Owen Lynch, Brandon T. Shapiro, and David I. Spivak. *All Concepts are* Cat\*. 2023. arXiv: 2305.02571 [math.CT] (cit. on p. 36).
- [Mar75] Per Martin-Löf. "An Intuitionistic Theory of Types: Predicative Part". In: *Logic Colloquium* '73. Ed. by H.E. Rose and J.C. Shepherdson. Vol. 80. Studies in Logic and the Foundations of Mathematics. Elsevier, 1975, pp. 73–118 (cit. on p. 1).
- [NS24] Nelson Niu and David I. Spivak. *Polynomial Functors: A Mathematical Theory of Interaction*. London Mathematical Society Lecture Notes, *to appear*. Cambridge University Press, 2024 (cit. on pp. 15, 44).

- [SS24] Brandon T. Shapiro and David I. Spivak. *A Polynomial Construction of Nerves for Higher Categories*. 2024. arXiv: 2405.13157 [math.CT]. url: https://arxiv.org/abs/2405.13157 (cit. on p. 36).
- [Voe+13] Vladimir Voevodsky et al. "Homotopy type theory: Univalent foundations of mathematics". In: *Institute for Advanced Study (Princeton), The Univalent Foundations Program* (2013) (cit. on pp. 2, 7, 10).
- [ZM22] Maaike Zwart and Dan Marsden. "No-go theorems for distributive laws". In: *Logical Methods in Computer Science* 18 (2022).

## 7 Appendix A

Some additional identities/theorems from HoTT used throughout this paper:

```
transpAp : \forall \{\ell \ \ell' \ \kappa\} \{A : Type \ \ell\} \{A' : Type \ \ell'\} \{a \ b : A\}
                  \rightarrow (B : A' \rightarrow Type \kappa) (f : A \rightarrow A') (e : a \equiv b) (x : B (f a))
                   \rightarrow transp (\lambda x \rightarrow B (f x)) e x \equiv transp B (ap f e) x
transpAp B f refl x = refl
•invr : \forall \{\ell\} {A : Type \ell} {a b : A}
             \rightarrow (e : a \equiv b) \rightarrow (sym e) \bullet e \equiv refl
•invr refl = refl
\equiv \text{siml} : \forall \{\ell\} \{A : \text{Type } \ell\} \{a b : A\}
             \rightarrow (e : a \equiv b) \rightarrow refl \equiv (b \equiv \langle sym e \rangle e)
\equivsiml refl = refl
\equiv idr : \forall \{\ell\} \{A : Type \ell\} \{a b : A\}
            \rightarrow (e : a \equiv b) \rightarrow e \equiv (a \equiv\langle refl \rangle e)
\equividr refl = refl
conj : \forall \{\ell\} \{A : Type \ell\} \{a b c d : A\}
           \rightarrow (e1 : a \equiv b) (e2 : a \equiv c) (e3 : b \equiv d) (e4 : c \equiv d)
           \rightarrow (a \equiv \left( e1 \rangle e3) \equiv (a \equiv \left\ e2 \rangle e4)
           \rightarrow e3 \equiv (b \equiv < sym e1 > (a \equiv < e2 > e4))
conj e1 e2 refl refl = ≡siml e1
nat : \forall \{\ell \ \kappa\} \{A : Type \ \ell\} \{B : Type \ \kappa\} \{f \ g : A \rightarrow B\} \{a \ b : A\}
          \rightarrow (\alpha: (x : A) \rightarrow f x \equiv g x) (e : a \equiv b)
          \rightarrow ((f a) \equiv \langle \alpha \ a \rangle (ap g e)) \equiv ((f a) \equiv \langle ap \ f \ e \rangle (\alpha \ b))
nat \{a = a\} \alpha refl = \equiv idr (\alpha \ a)
cancel : \forall \{\ell\} \{A : Type \ell\} \{a b c : A\}
               \rightarrow (e1 e2 : a \equiv b) (e3 : b \equiv c)
```

```
\rightarrow (a \equiv \langle e1 \rangle e3) \equiv (a \equiv \langle e2 \rangle e3)
                \rightarrow e1 \equiv e2
cancel e1 e2 refl refl = refl
apId : \forall \{\ell\} \{A : Type \ell\} \{a b : A\}
             \rightarrow (e : a \equiv b) \rightarrow ap (\lambda x \rightarrow x) e \equiv e
apId refl = refl
apComp : \forall {\ell \ell' \ell''} {A : Type \ell} {B : Type \ell'} {C : Type \ell''} {a b : A}
                \rightarrow (f : A \rightarrow B) (g : B \rightarrow C) (e : a \equiv b)
                \rightarrow ap (\lambda x \rightarrow g (f x)) e \equiv ap g (ap f e)
apComp f g refl = refl
apHtpy : \forall \{\ell\} {A : Type \ell} {a : A}
                \rightarrow (i : A \rightarrow A) (\alpha : (x : A) \rightarrow i x \equiv x)
                \rightarrow ap i (\alpha a) \equiv \alpha (i a)
apHtpy \{a = a\} i \alpha =
       cancel (ap i (\alpha a)) (\alpha (i a)) (\alpha a)
                    ((i (i a) \equiv \( ap i (\alpha a) \) \( \alpha \) a)
                    \equiv \langle \text{ sym (nat } \alpha \ (\alpha \ a)) \rangle
                    ((i (i a) \equiv \langle \alpha (i a) \rangle ap (\lambda z \rightarrow z) (\alpha a))
                    \equiv \langle ap (\lambda e \rightarrow i (i a) \equiv \langle \alpha (i a) \rangle e) (apId (\alpha a)) \rangle
                    ((i (i a) \equiv \langle \alpha (i a) \rangle \alpha a) \square)))
\mathsf{HAdj} : \forall \{\ell \ \kappa\} \ \{\mathsf{A} : \mathsf{Type} \ \ell\} \ \{\mathsf{B} : \mathsf{Type} \ \kappa\}
            \rightarrow (A \rightarrow B) \rightarrow Set (\ell \sqcup \kappa)
HAdj \{A = A\} \{B = B\} f =
       \Sigma (B \rightarrow A) (\lambda g \rightarrow
           \Sigma ((x : A) \rightarrow g (f x) \equiv x) (\lambda \eta \rightarrow
               \Sigma ((y : B) \rightarrow f (g y) \equiv y) (\lambda \epsilon \rightarrow
                  (x : A) \rightarrow ap f (\eta x) \equiv \epsilon (f x)))
Iso\rightarrowHAdj : \forall \{\ell \ \kappa\} \{A : Type \ \ell\} \{B : Type \ \kappa\} \{f : A \rightarrow B\}
                    → Iso f → HAdj f
Iso→HAdj {f = f} (g , \eta , \epsilon) =
       g, (\eta
        , ( (\lambda y \rightarrow
                    f(g y)
                                               \equiv \langle \text{ sym } (\epsilon \text{ (f (g y))}) \rangle
                   (f (g (f (g y))) \equiv \langle ap f (\eta (g y)) \rangle
                                                 \equiv \langle \epsilon y \rangle
                   (f (g y))
                   (y
                                                  □))))
           \lambda
                  \rightarrow conj (\epsilon (f (g (f x)))) (ap f (\eta (g (f x))))
                          (ap f (\eta x)) (\epsilon (f x))
```

```
(((f (g (f (g (f x)))) \equiv \langle \epsilon (f (g (f x))) \rangle ap f (\eta x)))
                       \equiv \langle \text{ nat } (\lambda z \rightarrow \epsilon (f z)) (\eta x) \rangle
                        (((f (g (f (g (f x)))))
                           \equiv \langle ap (\lambda z \rightarrow f (g (f z))) (\eta x) \rangle
                           \epsilon (f x)))
                       \equiv \langle ap (\lambda e \rightarrow (f (g (f (g (f x)))) \equiv \langle e \rangle \epsilon (f x)))
                                  ((ap (\lambda z \rightarrow f (g (f z))) (\eta x))
                                   \equiv \langle apComp (\lambda z \rightarrow g (f z)) f (\eta x) \rangle
                                   ((ap f (ap (\lambda z \rightarrow g (f z)) (\eta x)))
                                   \equiv \langle ap (ap f) (apHtpy (\lambda z \rightarrow g (f z)) \eta) \rangle
                                    (ap f (\eta (g (f x))) \square))) \rangle
                        (((f (g (f (g (f x))))
                           \equiv \langle \text{ ap f } (\eta (g (f x))) \rangle
                           \epsilon (f x))) \square)))))
pairEquiv1 : \forall {\ell \ell' \kappa} {A : Type \ell} {A' : Type \ell'} {B : A' \rightarrow Type \kappa}
                      \rightarrow (f : A \rightarrow A') \rightarrow isEquiv f
                      \rightarrow isEquiv {A = \Sigma A (\lambda x \rightarrow B (f x))} {B = \Sigma A' B}
                                       (\lambda (x, y) \rightarrow (f x, y))
pairEquiv1 \{A = A\} \{A' = A'\} \{B = B\} f ef =
   Iso→isEquiv
       ((\lambda (x, y) \rightarrow (g x, transp B (sym (\epsilon x)) y))
       , ( (\lambda (x, y) \rightarrow pairEq (\eta x) (lemma x y))
          , \lambda (x , y) \rightarrow pairEq (\epsilon x) (symr (\epsilon x) y) )
   where
      g : A' \rightarrow A
      g = fst (Iso \rightarrow HAdj (isEquiv \rightarrow Iso ef))
      \eta : (x : A) \rightarrow g (f x) \equiv x
      \eta = \text{fst (snd (Iso}\rightarrow \text{HAdj (isEquiv}\rightarrow \text{Iso ef)))}
      \epsilon: (y : A') \rightarrow f (g y) \equiv y
      \epsilon = fst (snd (snd (Iso\rightarrowHAdj (isEquiv\rightarrowIso ef))))
      \rho : (x : A) \rightarrow ap f (\eta x) \equiv \epsilon (f x)
       \rho = \text{snd} (\text{snd} (\text{snd} (\text{Iso}\rightarrow \text{HAdj} (\text{isEquiv}\rightarrow \text{Iso ef}))))
      lemma : (x : A) (y : B (f x))
                    \rightarrow transp (\lambda z \rightarrow B (f z)) (\eta x)
                                    (transp B (sym (\epsilon (f x))) y)
                        ≡ y
      lemma x y = (transp (\lambda z \rightarrow B (f z)) (\eta x)
                                        (transp B (sym (\epsilon (f x))) y))
                           \equiv \langle \text{ transpAp B f } (\eta x) \rangle
                                               (transp B (sym (\epsilon (f x))) y) \rangle
                           ( transp B (ap f (\eta x))
                                              (transp B (sym (\epsilon (f x))) y)
                           \equiv \langle ap (\lambda e \rightarrow transp B e
```

```
(transp B (sym (\epsilon (f x))) y))
                                  (\rho x)
                         ( (transp B (\epsilon (f x))
                                         (transp B (sym (\epsilon (f x))) y))
                         \equiv \langle (\text{symr} (\epsilon (f x)) y) \rangle
                         (y □)))
pairEquiv2 : \forall \{\ell \ \kappa \ \kappa'\} \{A : Type \ \ell\}
                    \rightarrow {B : A \rightarrow Type \kappa} {B' : A \rightarrow Type \kappa'}
                    \rightarrow (g : (x : A) \rightarrow B x \rightarrow B' x) \rightarrow ((x : A) \rightarrow isEquiv (g x))
                    \rightarrow isEquiv {A = \Sigma A B} {B = \Sigma A B'}
                                    (\lambda (x, y) \rightarrow (x, g x y))
pairEquiv2 g eg =
      let isog = (\lambda \times \rightarrow isEquiv \rightarrow Iso (eg \times))
      in Iso\rightarrowisEquiv ( (\lambda (x , y) \rightarrow (x , fst (isog x) y))
                              , ( (\lambda (x, y) \rightarrow
                                        pairEq refl (fst (snd (isog x)) y))
                                 \lambda (x , y) \rightarrow
                                       pairEq refl (snd (snd (isog x)) y)))
pairEquiv : \forall \{\ell \ \ell' \ \kappa \ \kappa'\} \{A : Type \ \ell\} \{A' : Type \ \ell'\}
                  \rightarrow {B : A \rightarrow Type \kappa} {B' : A' \rightarrow Type \kappa'}
                  \rightarrow (f : A \rightarrow A') (g : (x : A) \rightarrow B x \rightarrow B' (f x))
                  \rightarrow isEquiv f \rightarrow ((x : A) \rightarrow isEquiv (g x))
                  \rightarrow isEquiv {A = \Sigma A B} {B = \Sigma A' B'}
                                   (\lambda (x, y) \rightarrow (f x, g x y))
pairEquiv f g ef eg =
      compIsEquiv (pairEquiv1 f ef)
                         (pairEquiv2 g eg)
J: \forall \{\ell \ \kappa\} \{A: Type \ \ell\} \{a: A\} (B: (x:A) \rightarrow a \equiv x \rightarrow Type \ \kappa)
      \rightarrow {a' : A} (e : a \equiv a') \rightarrow B a refl \rightarrow B a' e
J B refl b = b
transpPre : \forall \{\ell 0 \ \ell 1 \ \kappa 0 \ \kappa 1\} \{A : Type \ \ell 0\} \{a \ a' : A\} \{B : A \rightarrow Type \ \kappa 0\}
                      {C : Type \ell 1} {D : C \rightarrow Type \kappa 1} {f : A \rightarrow C}
                      (mf : isMono f) (g : (x : A) \rightarrow B x \rightarrow D (f x))
                      (e : f a \equiv f a') \{b : B a\}
                      \rightarrow transp D e (g a b) \equiv g a' (transp B (inv mf e) b)
transpPre \{a = a\} \{a' = a'\} \{B = B\} \{D = D\} \{f = f\} mf g e \{b = b\} = f\}
      transp D e (g a b)
            \equiv \langle ap (\lambda e' \rightarrow transp D e' (g a b)) (sym (snd (snd mf) e)) \rangle
      (\_ \equiv \langle (J (\lambda x e' \rightarrow transp D (ap f e') (g a b) \equiv g x (transp B e' b))
                       (inv mf e) refl) >
```

```
((g a' (transp B (inv mf e) b)) \square))
postulate
       funext : \forall \{\ell \ \kappa\} \{A : Type \ \ell\}
                     \rightarrow {B : A \rightarrow Type \kappa} {f g : (x : A) \rightarrow B x}
                     \rightarrow ((x : A) \rightarrow f x \equiv g x) \rightarrow f \equiv g
       funextr : \forall \{\ell \ \kappa\} \ \{A : Type \ \ell\}
                       \rightarrow {B : A \rightarrow Type \kappa} {f g : (x : A) \rightarrow B x}
                       \rightarrow (e : (x : A) \rightarrow f x \equiv g x) \rightarrow coAp (funext e) \equiv e
       funextl : \forall \{\ell \ \kappa\} \{A : Type \ \ell\}
                       \rightarrow {B : A \rightarrow Type \kappa} {f g : (x : A) \rightarrow B x}
                       \rightarrow (e : f \equiv g) \rightarrow funext (coAp e) \equiv e
transpD : \forall \{\ell \ \kappa\} \{A : Type \ \ell\} \{B : A \rightarrow Type \ \kappa\} \{a \ a' : A\}
                \rightarrow (f : (x : A) \rightarrow B x) (e : a \equiv a')
                \rightarrow transp B e (f a) \equiv f a'
transpD f refl = refl
transpHAdj : \forall \{\ell \ \ell' \ \kappa\} \{A : Type \ \ell\} \{B : Type \ \ell'\}
                    \rightarrow {C : B \rightarrow Type \kappa} {a : A}
                   \rightarrow \{g : A \rightarrow B\} \{h : B \rightarrow A\}
                    \rightarrow (f: (x : A) \rightarrow C (g x))
                    \rightarrow (e : (y : B) \rightarrow g (h y) \equiv y)
                    \rightarrow (e': (x : A) \rightarrow h (g x) \equiv x)
                    \rightarrow (e'' : (x : A) \rightarrow e (g x) \equiv ap g (e' x))
                    \rightarrow transp C (e (g a)) (f (h (g a))) \equiv f a
transpHAdj \{C = C\} \{a = a\} \{g = g\} \{h = h\} f e e' e'' =
      transp C (e (g a)) (f (h (g a)))
             \equiv \langle ap (\lambda ee \rightarrow transp C ee (f (h (g a)))) (e'' a) \rangle
       (transp C (ap g (e' a)) (f (h (g a)))
             \equiv \langle \text{ sym (transpAp C g (e' a) (f (h (g a))))} \rangle
       ((transp (\lambda x \rightarrow C (g x)) (e' a) (f (h (g a))))
             \equiv \langle \text{ transpD f (e' a)} \rangle
       ((f a) \square))
PreCompEquiv : \forall \{\ell \ \ell' \ \kappa\} \ \{A : Type \ \ell\}
                         \rightarrow {B : Type \ell'} {C : B \rightarrow Type \kappa}
                         \rightarrow (f : A \rightarrow B) \rightarrow isEquiv f
                         \rightarrow isEquiv {A = (b : B) \rightarrow C b}
                                          {B = (a : A) \rightarrow C (f a)}
                                          (\lambda \ g \rightarrow \lambda \ a \rightarrow g \ (f \ a))
PreCompEquiv {C = C} f ef =
      let (f^{-1}, 1, r, e) = Iso \rightarrow HAdj (is Equiv \rightarrow Iso ef)
      in Iso\rightarrowisEquiv ( (\lambda g b \rightarrow transp C (r b) (g (f<sup>-1</sup> b)))
```